

SAS[®] 9.3 データセットオプション リファレンス

The correct bibliographic citation for this manual is as follows: SAS Institute Inc. 2011. *SAS® 9.3 のデータセットオプション: リファレンス*. Cary, NC: SAS Institute Inc.

SAS® 9.3 のデータセットオプション: リファレンス

Copyright © 2011, SAS Institute Inc., Cary, NC, USA

ISBN 978-1-60764-899-4 (electronic book)

All rights reserved. Produced in the United States of America.

For a hardcopy book: No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

For a Web download or e-book: Your use of this publication shall be governed by the terms established by the vendor at the time you acquire this publication.

The scanning, uploading, and distribution of this book via the Internet or any other means without the permission of the publisher is illegal and punishable by law. Please purchase only authorized electronic editions and do not participate in or encourage electronic piracy of copyrighted materials. Your support of others' rights is appreciated.

U.S. Government Restricted Rights Notice: Use, duplication, or disclosure of this software and related documentation by the U.S. government is subject to the Agreement with SAS Institute and the restrictions set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

Printing 1, 2011 July

SAS® Publishing provides a complete selection of books and electronic products to help customers use SAS software to its fullest potential. For more information about our e-books, e-learning products, CDs, and hard-copy books, visit the SAS Publishing Web site at support.sas.com/publishing or call 1-800-727-3228.

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

目次

本書について.....	v
SAS 9.3 データセットオプションの新機能.....	ix
推奨資料.....	xi
1 章・概念.....	1
ユーザー補助.....	1
データセットオプションについて.....	1
構文.....	2
データセットオプションの使用.....	2
他の SAS ドキュメントで説明されているデータセットオプション.....	3
2 章・データセットオプションリファレンス.....	5
カテゴリ別データセットオプション.....	6
ディクショナリ.....	8
キーワード.....	71

本書について

SAS 言語の構文規則

SAS 言語の構文規則の概要

SAS 言語要素の構文のドキュメントでは、標準の規則が使用されています。この規則により、SAS 構文のコンポーネントを簡単に識別できます。規則は次の 3 項目に分けることができます。

- 構文のコンポーネント
- スタイル規則
- 特殊文字
- SAS ライブラリと外部ファイルへの参照

構文のコンポーネント

ほとんどの言語要素の構文のコンポーネントには、キーワードと引数があります。キーワードのみが必要な言語要素もあります。その他の言語要素では、キーワードの後に等号記号(=)が付きます。

キーワード

プログラムを記述するときに使用する SAS 言語要素の名前を指定します。キーワードはリテラルで、通常は構文の最初の単語です。CALL ルーチンでは、最初の 2 つの単語がキーワードです。

次の SAS 構文の例では、キーワードは構文の最初の単語です。

```
CHAR (string, position)
CALL RANBIN (seed, n, p, x);
ALTER (alter-password)
BEST w.
REMOVE <data-set-name>
```

次の例では、CALL ルーチンの最初の 2 つの単語がキーワードです。

```
CALL RANBIN(seed, n, p, x)
```

一部の SAS ステートメントの構文は、引数のない 1 つのキーワードで構成されません。

```
DO;
... SAS code ...
```

END;

一部のシステムオプションでは、2 つあるキーワード値から 1 つを指定する必要があります。

DUPLEX | NODUPLEX

引数

数値または文字の定数、変数、式のいずれかを指定します。引数は、キーワードまたはキーワードの後の等号記号の後に続きます。SAS では、言語要素を処理するために引数が使用されます。引数は必須の場合と、省略可能な場合があります。構文中では、省略可能な引数は山かっこで囲まれます。

次の例では、*string* と *position* がキーワード CHAR の後に続きます。これらの引数は、CHAR 関数の必須の引数です。

CHAR (*string*, *position*)

各引数には値があります。次の SAS コードの例では、引数 *string* の値は 'summer'、引数 *position* の値は 4 です。x=char('summer', 4);

次の例では、*string* と *substring* が必須の引数で、*modifiers* と *startpos* は省略可能な引数です。

FIND(*string*, *substring* <,*modifiers*> <,*startpos*>

注: ほとんどの場合、SAS ドキュメント内のコード例は、等幅フォントを使用して小文字で記述されています。コードを記述するときは、大文字のみ、小文字のみ、または大文字と小文字の両方を使用できます。

スタイル規則

SAS 構文のドキュメントで使用されているスタイル規則には、太字の大文字、小文字、斜体があります。

太字の大文字

関数またはステートメントの名前など、SAS キーワードを示します。次の例では、キーワード ERROR が太字の大文字で記述されています。

ERROR<message>;

大文字

リテラルである引数を示します。

次の CMPMODEL=システムオプションの例には、BOTH、CATALOG、XML というリテラルが含まれています。

CMPMODEL = BOTH | CATALOG | XML

斜体

ユーザーが指定する引数または値を示します。斜体の項目は、ユーザーが指定する次のいずれかの値を表します。

- 非リテラル引数

次の LINK ステートメントの例では、引数 *label* はユーザーが指定する値であるため、斜体で記述されています。

LINK *label*;

- 引数に割り当てる非リテラル値

次の FORMAT ステートメントの例では、引数 DEFAULT には変数 *default-format* が割り当てられます。

FORMAT = *variable-1* <, ..., *variable-n format* ><DEFAULT = *default-format*>;

斜体の項目は、選択可能な引数リストの一般名 (*attribute-list* など) である場合もあります。斜体の項目を複数使用する場合、項目は *item-1*, ..., *item-n* のように表されます。

特殊文字

SAS 言語要素の構文には、次の特殊文字を使用できます。

=

等号記号は、システムオプションなどの一部の言語要素内のリテラルの値を示します。

次の MAPS システムオプションの例では、等号記号で MAPS の値が設定されます。

MAPS = *location-of-maps*

<>

山かっこは省略可能な引数を示します。山かっこで囲まれていない引数は必須です。

次の CAT 関数の例では、少なくとも 1 つの項目が必須です。

CAT (*item-1* <, ..., *item-n*>)

縦棒は一連の値の中から 1 つ選択できることを示します。縦棒で区切られている値は相互に排他的です。

次の CMPMODEL=システムオプションの例では、引数を 1 つのみ選択できます。

CMPMODEL = BOTH | CATALOG | XML

...

省略記号は、省略記号の後に続く引数または一連の引数を繰り返せることを示します。省略記号とそれに続く引数が山かっこで囲まれている場合、その引数は省略可能です。

次の CAT 関数の例では、省略記号は複数の省略可能な項目を指定できることを示します。

CAT (*item-1* <, ..., *item-n*>)

'value'または" value"

一重または二重引用符で囲まれた引数には、同じく一重または二重引用符で囲まれた値を指定する必要があることを示します。

次の FOOTNOTE ステートメントの例では、引数 *text* が引用符で囲まれています。

FOOTNOTE <*n*> <*ods-format-options* '*text*' | "*text*">;

;

セミコロンは、ステートメントまたは CALL ルーチンの終了を示します。

次の例では、各ステートメントがセミコロンで終了しています。data namegame;
length color name \$8; color = 'black'; name = 'jack'; game =
trim(color) || name; run;

SAS ライブラリと外部ファイルへの参照

SAS ステートメントおよびその他の言語要素の多くは、SAS ライブラリと外部ファイルを参照します。論理名(ライブラリ参照名またはファイル参照名)を使用して参照するか、引用符で囲んだ物理ファイル名を使用するかを選択できます。論理名を使用する場合は、通常、SAS ステートメント(LIBNAME または FILENAME)を使用するか、動作環境の制御言語を使用して関連付けをするかを選択できます。SAS ライブラリと外部ファイルを参照する方法は複数あり、一部の方法は動作環境によって異なります。

SAS ドキュメント中の外部ファイルを使用する例では、*file-specification* という語句を斜体で使用しています。SAS ドキュメント中の SAS ライブラリを使用する例では、*SAS-library* という語句を斜体で使用しています。*SAS-library* は次のように引用符で囲まれます。

```
infile file-specification obs = 100;  
libname libref 'SAS-library';
```


SAS 9.3 データセットオプションの新機能

概要

SAS データセットオプションのドキュメントは、現在 *SAS Language Reference: Dictionary* には含まれていません。*SAS Language Reference: Dictionary* に以前記載されていた SAS データセットオプションは、現在この *SAS 9.3 データセットオプション: リファレンス* に記載されています。

SAS Language Reference: Dictionary への変更

SAS 9.3 より前は、このドキュメントは *SAS Language Reference: Dictionary* に含まれていました。SAS 9.3 以降では、*SAS Language Reference: Dictionary* は 7 つのドキュメントに分割されています。

- *SAS データセットオプション: リファレンス*
- *SAS 出力形式と入力形式: リファレンス*
- *SAS 関数と CALL ルーチン: リファレンス*
- *SAS ステートメント: リファレンス*
- *SAS システムオプション: リファレンス*
- *SAS コンポーネントオブジェクト: リファレンス* (ハッシュオブジェクトおよび Java オブジェクトのドキュメントを含む)
- *Base SAS Utilities: リファレンス* (SAS DATA ステップデバグおよび SAS ユーティリティマクロ%DS2CSV を含む)

新しい SAS データセットオプション

新しいデータセットオプションを次に示します。

[EXTENDOBSCOUNTER=](#) (p. 19)

出力 SAS データファイルの最大オブザベーションカウントを拡張します。

x SAS データセットオプション

推奨資料

- *SAS 言語リファレンス: 解説編*
- *Base SAS Glossary*
- *An Array of Challenges - Test Your SAS Skills*
- *Base SAS プロシジャガイド*
- *Combining and Modifying SAS Data Sets: Examples*
- *The Little SAS Book: A Primer*
- *SAS Programming by Example*
- *Step-by-Step Programming with Base SAS Software*
- *Using the SAS Windowing Environment: A Quick Tutorial*

SAS の刊行物の総一覧については、support.sas.com/bookstore にてご確認ください。
必要な書籍についてのご質問は、下記までお寄せください。

SAS Publishing Sales
SAS Campus Drive
Cary, NC 27513-2414
電話: 1-800-727-3228
ファクシミリ: 1-919-677-8166
電子メール: sasbook@sas.com
Web アドレス: support.sas.com/bookstore

1 章 概念

ユーザー補助	1
データセットオプションについて	1
構文	2
データセットオプションの使用	2
入力または出力 SAS データセットにデータセットオプションを使用する	2
データセットオプションとシステムオプションの相互作用	3
他の SAS ドキュメントで説明されているデータセットオプション	3

ユーザー補助

このドキュメントは、コマンドベースの製品です。このリリースで、ユーザー補助に対応するために追加された機能はありませんが、この製品はもともとグラフィカルユーザーインターフェイスがなく、キーボードを打つことができるか、もしくは他の方法でコマンドを入力できればすべての機能を利用できるため、ユーザー補助の標準に準拠している可能性が高いと考えています。SAS 製品のユーザー補助機能についてのご質問は、accessibility@sas.com まで電子メールでお寄せいただくか、SAS テクニカルサポートにお問い合わせください。

データセットオプションについて

データセットオプションは、データセットオプションが表示される SAS データセットにのみ適用されるアクションを指定します。次の操作を実行できます。

- 変数名の変更
- 最初または最後の n 個のオブザベーションのみを処理対象として選択する
- 処理または出力データセットから変数を削除する
- データセットのパスワードを指定する

構文

SAS データセット名の後に、データセットオプションをカッコで囲んで指定します。複数のデータセットオプションを指定するには、空白で区切ります。

(option-1=value-1<...option-n=value-n>)

これらの例では、SAS ステートメント内のデータセットオプションを示します。

- data scores(keep=team game1 game2 game3);
- data mydata(index=(b k) label='label for my data set' drop=p read=secret);
- data new(drop=i n index=(j combo=(x1 a1 a20 b1 b50)));
- data idxdup2(compress=yes index=(ok1 ok2 ssn/unique ok3));
- proc print data=new(drop=year);
- set old(rename=(date=Start_Date));

データセットオプションの使用

入力または出力 SAS データセットにデータセットオプションを使用する

ほとんどの SAS データセットオプションは、DATA ステップまたはプロシジャ(PROC)ステップ内の入力または出力 SAS データセットに適用できます。データセットオプションが入力データセットに関連付けられている場合、アクションは読み込まれているデータセットに適用されます。オプションが DATA ステートメント内に表示されるか、PROC ステップ内で出力データセットの指定の後に表示されると、アクションは出力データセットに適用されます。DATA ステップでは、出力データセットのデータセットオプションは、存在する可能性がある OUTPUT ステートメント内であればよいのではなく、DATA ステートメント内に表示される必要があります。

COMPRESS=などの一部のデータセットオプションは、そのデータセットの存続期間に存在する属性を設定するため、SAS データセットを作成するときのみ意味があります。多数のデータセットオプションを変更またはキャンセルするには、データセットを再作成する必要があります。他のオプション(PW=や LABEL=など)は PROC DATASETS を使用して変更できます。詳細については、“DATASETS プロシジャ”(Base SAS プロシジャガイド)を参照してください。

データセットオプションが同じ DATA または PROC ステップ内の入力データセットと出力データセットの両方に表示される場合、データセットオプションは先に入力データセットに適用されます。次に、プログラミングステートメントが評価されるか、データセットオプションが出力データセットに適用されます。同様に、作成しているデータセットに指定されるデータセットオプションは、プログラミングステートメントが処理された後に適用されます。たとえば、RENAME=データセットオプションを使用しているときには、新しい名前は DATA ステップが終了するまで変数に関連付けられません。

ときには、同じステートメント内で使用されているデータセットオプションが競合する場合があります。たとえば、同じステートメント内の同じ変数に DROP=データセットオプションと KEEP=データセットオプションの両方は指定できません。また、タイミングが問題になる場合もあります。たとえば、SET ステートメントで指定されたデータセットに対し KEEP=と RENAME=を使用する場合、KEEP=では元の変数名を使用する必要が

あります。KEEP=はデータセットが読み込まれる前に処理されます。RENAME=で指定した新しい名前は、SET ステートメントの後のプログラミングステートメントに適用されます。

データセットオプションとシステムオプションの相互作用

システムオプションおよびデータセットオプションの多くは、同じ名前を共有し、同じ関数を使用します。システムオプションは、SAS ジョブまたはセッション内のすべての DATA および PROC ステップに対して有効な状態を保ちます。

データセットオプションは、表示されるステップ内のデータセットのシステムオプションよりも優先されます。この例では、OPTIONS ステートメント内の OBS=システムオプションが、SAS ジョブ内のデータセットから最初の 100 個のオブザベーションのみが処理されることを指定しています。ただし、SET ステートメント内の OBS=データオプションが、データセット TWO のシステムオプションよりも優先され、データセット TWO から最初の 5 個のオブザベーションのみを読み込むように指定されます。PROC PRINT ステップによりデータセット FINAL が出力されます。このデータセットには、データセット TWO の最初の 5 個のオブザベーション、次にデータセット THREE の最初の 100 個のオブザベーションが含まれます。

```
options obs=100;

data final;
set two(obs=5) three;
run;

proc print data=final;
run;
```

他の SAS ドキュメントで説明されているデータセットオプション

SAS Language Reference: Dictionary に記載されているデータセットオプションに加えて、次のドキュメントにもデータセットオプションが記載されています。

- *SAS Companion for Windows*
- *SAS Companion for UNIX Environments*
- *SAS Companion for z/OS*
- *SAS National Language Support: Reference Guide*
- *SAS Scalable Performance Data Engine: Reference*
- *SAS/ACCESS for Relational Databases: References*

2 章

データセットオプションリファレンス

カテゴリ別データセットオプション	6
ディクショナリ	8
ALTER=データセットオプション	8
BUFNO=データセットオプション	9
BUFSIZE=データセットオプション	11
CNTLLEV=データセットオプション	12
COMPRESS=データセットオプション	13
DLDMGACTION=データセットオプション	15
DROP=データセットオプション	16
ENCRYPT=データセットオプション	17
EXTENDOBSCOUNTER=データセットオプション	19
FILECLOSE=データセットオプション	20
FIRSTOBS=データセットオプション	20
GENMAX=データセットオプション	22
GENNUM=データセットオプション	23
IDXNAME=データセットオプション	25
IDXWHERE=データセットオプション	26
INDEX=データセットオプション	28
IN=データセットオプション	29
KEEP=データセットオプション	30
LABEL=データセットオプション	31
OBSBUF=データセットオプション	33
OBS=データセットオプション	35
OUTREP=データセットオプション	44
POINTOBS=データセットオプション	45
PW=データセットオプション	47
PWREQ=データセットオプション	48
READ=データセットオプション	48
RENAME=データセットオプション	49
REPEMPTY=データセットオプション	51
REPLACE=データセットオプション	52
REUSE=データセットオプション	53
ROLE=データセットオプション	54
SORTEDBY=データセットオプション	55
SPILL=データセットオプション	57
TOBSNO=データセットオプション	65
TYPE=データセットオプション	65
WHERE=データセットオプション	66
WHEREUP=データセットオプション	68
WRITE=データセットオプション	69

カテゴリ別データセットオプション

SAS データセットオプションのカテゴリは、SAS データセットオプショングループに対応します。

データセットコントロール	データセットに関連付けられたオプション
オブザベーションコントロール	オブザベーションに関連付けられたオプション
SAS インデックス使用のユーザーコントロール	インデックスに関連付けられたオプション
変数制御	変数に関連付けられたオプション
その他	

カテゴリ	言語要素	説明
SAS インデックス使用のユーザーコントロール	IDXNAME=データセットオプション (p. 25)	WHERE 式の条件の照合に特定のインデックスを使用するように SAS に指示します。
	IDXWHERE=データセットオプション (p. 26)	SAS で WHERE 式の条件の照合にインデックス検索を使用するか、順次検索を使用するかを指定します。
オブザベーションコントロール	FIRSTOBS=データセットオプション (p. 20)	SAS データセット内で SAS が最初に処理するオブザベーションを指定します。
	IN=データセットオプション (p. 29)	データセットのデータが現在のオブザベーションに関与しているかどうかを示すブール値変数を作成します。
	OBS=データセットオプション (p. 35)	SAS が処理するデータセット内の最後のオブザベーションを指定します。
	POINTOBS=データセットオプション (p. 45)	SAS でオブザベーションへのランダムアクセスまたは順次アクセスが可能な圧縮データセットを作成するかどうかを指定します。
	WHERE=データセットオプション (p. 66)	SAS データセットからオブザベーションを選択するために使用する特定の条件を指定します。
	WHEREUP=データセットオプション (p. 68)	新しいオブザベーションと変更されたオブザベーションを WHERE 式に対して評価するかどうかを指定します。
その他	FILECLOSE=データセットオプション (p. 20)	SAS データセットを閉じたときのテープの位置付けを指定します。
データセットコントロール	ALTER=データセットオプション (p. 8)	ALTER パスワードを SAS ファイルに割り当てて、ユーザーによるファイルの置き換えや削除を防止し、読み取り保護や書き込み保護されたファイルへのアクセスを可能にします。
	BUFNO=データセットオプション (p. 9)	SAS データセットの処理用に割り当てるバッファ数を指定します。

カテゴリ	言語要素	説明
	BUFSIZE=データセットオプション (p. 11)	出力 SAS データセット用の永久バッファページのサイズを指定します。
	CNTLLEV=データセットオプション (p. 12)	SAS データセットへの共有アクセスのレベルを指定します。
	COMPRESS=データセットオプション (p. 13)	新しい出力 SAS データセットでのオブザベーションの圧縮方法を指定します。
	DLDMGACTION=データセットオプション (p. 15)	SAS ライブラリ内の SAS データセットの損傷が検出されたときに実行するアクションを指定します。
	ENCRYPT=データセットオプション (p. 17)	出力 SAS データセットを暗号化するかどうかを指定します。
	EXTENDOBSCOUNTER=データセットオプション (p. 19)	新しい出力 SAS データファイル内の最大オブザベーションカウントを拡張するかどうかを指定します。
	GENMAX=データセットオプション (p. 22)	新しいデータセットの世代を要求し、既存のデータセットの世代数を変更し、バージョンの最大数を指定します。
	GENNUM=データセットオプション (p. 23)	SAS データセットの特定の世代を指定します。
	INDEX=データセットオプション (p. 28)	新しい出力 SAS データセットのインデックスを定義します。
	LABEL=データセットオプション (p. 31)	SAS データセットのラベルを指定します。
	OBSBUF=データセットオプション (p. 33)	DATA ステップビューを処理する表示バッファのサイズを決定します。
	OUTREP=データセットオプション (p. 44)	出力 SAS データセットのデータ表記を指定します。
	PW=データセットオプション (p. 47)	SAS ファイルに READ、WRITE および ALTER パスワードを割り当て、パスワード保護された SAS ファイルへのアクセスを可能にします。
	PWREQ=データセットオプション (p. 48)	SAS データセットパスワードを入力するダイアログボックスを表示するかどうかを指定します。
	READ=データセットオプション (p. 48)	READ パスワードを SAS ファイルに割り当てます。これにより、ユーザーはパスワードを入力しない限りファイルを読み取れなくなります。
	REPEMPTY=データセットオプション (p. 51)	同じ名前の新しい空のデータセットで、既存の SAS データセットを上書きできるかどうかを指定します。
	REPLACE=データセットオプション (p. 52)	データが含まれている同じ名前の新しい SAS データセットで、既存のデータセットを上書きできるかどうかを指定します。

カテゴリ	言語要素	説明
	REUSE=データセットオプション (p. 53)	圧縮 SAS データセットの空き領域に新しいオブザベーションを書き込みできるかどうかを指定します。
	ROLE=データセットオプション (p. 54)	スタースキーマ結合のファクトテーブルを指定します。
	SORTEDBY=データセットオプション (p. 55)	データセットの現在の並べ替え方法を指定します。
	SPILL=データセットオプション (p. 57)	DATA ステップビューの非順次処理の予備ファイルを作成するかどうかを指定します。
	TOBSNO=データセットオプション (p. 65)	クライアント/サーバー間の転送で送信するオブザベーション数を指定します。
	TYPE=データセットオプション (p. 65)	特別構造の SAS データセットにデータセットの種類を指定します。
	WRITE=データセットオプション (p. 69)	WRITE パスワードを SAS ファイルに割り当てます。これにより、ユーザーはパスワードを入力しない限りファイルを書き込めなくなります。
変数制御	DROP=データセットオプション (p. 16)	入力データセットの場合は、指定された変数を処理対象から除外し、出力データセットの場合は、指定された変数をデータセットへの書き込み対象から除外します。
	KEEP=データセットオプション (p. 30)	入力データセットの場合は、処理する変数を指定し、出力データセットの場合は、データセットに書き込む変数を指定します。
	RENAME=データセットオプション (p. 49)	変数の名前を変更します。

ディクショナリ

ALTER=データセットオプション

ALTER パスワードを SAS ファイルに割り当てて、ユーザーによるファイルの置き換えや削除を防止し、読み取り保護や書き込み保護されたファイルへのアクセスを可能にします。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

参照項目: 動作環境向けドキュメントの OpenVMS、UNIX または z/OS の下にある ALTER=データセットオプション。

構文

`ALTER=alter-password`

構文の説明

alter-password

有効な SAS 名にする必要があります。Rules for Words and Names in the SAS Language を参照してください。

詳細

ALTER=オプションは、カタログ以外のすべての種類の SAS ファイルに適用されません。このオプションは、パスワードを SAS ファイルに割り当てる場合や、読み取り保護、書き込み保護、または変更保護された SAS ファイルにアクセスする場合に使用できます。

ALTER パスワードで保護された SAS データセットを置き換えると、新しいデータセットに ALTER パスワードが継承されます。新しいデータセットの ALTER パスワードを変更するには、DATASETS プロシジャで MODIFY ステートメントを使用します。

注: SAS パスワードでは、SAS システム以外での SAS ファイルへのアクセスはコントロールされません。SAS 以外で SAS ファイルへのアクセスをコントロールするには、オペレーティングシステムで提供されるユーティリティか、ファイルシステムのセキュリティコントロールを使用する必要があります。

関連項目:

データセットオプション:

- “ENCRYPT=データセットオプション” (17 ページ)
- “PW=データセットオプション” (47 ページ)
- “READ=データセットオプション” (48 ページ)
- “WRITE=データセットオプション” (69 ページ)

その他:

- “ファイルの保護” (SAS 言語リファレンス: 解説編 34 章)
- “パスワードの操作” (Base SAS プロシジャガイド 15 章)

BUFNO=データセットオプション

SAS データセットの処理用に割り当てるバッファ数を指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

参照項目: 動作環境向けドキュメントの BUFNO=データセットオプション。

構文

BUFNO= *n* | *nK* | *hexX* | MIN | MAX

構文の説明

n | *nK*

バッファ数を 1(バイト)または 1,024(キロバイト)の倍数で指定します。たとえば、値 8 では 8 個のバッファ、値 1k では 1024 個のバッファが指定されます。

hexX

バッファ数を 16 進値で指定します。先頭が数値(0~9)、末尾が X の値を指定する必要があります。たとえば、値 2dx ではバッファ数が 45 バッファに設定されます。

MIN

最小バッファ数を 0 に設定します。これにより SAS では動作環境に最適な最小値が使用されます。これがデフォルト設定です。

MAX

バッファ数を動作環境で可能な最大数に設定します。4 バイト符号付き整数の最大値である $2^{31}-1$ (約 20 億)以下の値になります。

詳細

バッファ数は、データセットの永続的の属性ではなく、現在の SAS セッションまたはジョブでのみ有効です。

BUFNO=は、入力、出力または更新用に開かれている SAS データセットに適用されません。

バッファ数を大きくするほど、特定の SAS データセットに必要な入力および出力(I/O)操作の数を制限して、実行時間を短縮できます。ただし、実行時間が改善するかわりにメモリ消費が増えます。

小さいデータセットに対する I/O 操作を減らすとともに実行時間をスピードアップするには、処理するデータのページごとに 1 個のバッファを割り当てます。この手法は、処理中に同じオブザベーションを複数回読み込むときに最も効果的です。

動作環境の情報

BUFNO=のデフォルト値は、動作環境に応じて決まり、順次アクセスを最適化するように設定されます。直接(ランダム)アクセスの処理速度を向上させるには、BUFNO=の値を変更する必要があります。直接アクセスのデフォルト設定と使用可能な設定については、動作環境向け SAS ドキュメントの BUFNO=データセットオプションを参照してください。

比較

- BUFNO=データセットオプションが指定されていない場合、BUFNO=システムオプションの値が使用されます。同じ SAS セッションで両方の値が指定されている場合、BUFNO=データセットオプションに指定された値は、BUFNO=システムオプションに指定された値よりも優先されます。
- SAS でデータセットページとインデックスファイルページの数に基づいてバッファ数が割り当てられるように要求するには、SASFILE グローバルステートメントを使用します。

関連項目:**データセットオプション:**

- [“BUFSIZE=データセットオプション” \(11 ページ\)](#)

システムオプション:

- [“BUFNO=システムオプション” \(SAS システムオプション: リファレンス\)](#)

ステートメント:

- [“SASFILE ステートメント” \(SAS ステートメント: リファレンス\)](#)

BUFSIZE=データセットオプション

出力 SAS データセット用の永久バッファページのサイズを指定します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	データセットコントロール
制限事項:	出力データセットにのみ使用します。
参照項目:	動作環境向けドキュメントの UNIX、z/OS または OpenVMS の下にある BUFSIZE=データセットオプション。

構文

BUFSIZE= *n* | *nK* | *nM* | *nG* | *hexX* | MAX

構文の説明

n* | *nK* | *nM* | *nG

ページサイズを 1(バイト)、1,024(キロバイト)、1,048,576(メガバイト)、1,073,741,824(ギガバイト)のいずれかの倍数で指定します。たとえば、値 8 では 8 バイトのページサイズ、値 4k では 4096 バイトのページサイズが指定されます。

注: システムオプションとデータセットオプションのどちらも指定されていない場合、デフォルトは 0 です。その結果、動作環境に最適な最小ページサイズが使用されます。次のいずれかの場合は BUFSIZE=システムオプションが使用されます。

- BUFSIZE=データセットオプションが設定されていない
- BUFSIZE=データセットオプションがゼロに設定されている

バッファページサイズを動作環境のデフォルト値にリセットするには、BUFSIZE=0 を使用します。

hexX

ページサイズを 16 進値で指定します。先頭が数値(0~9)、末尾が X の値を指定する必要があります。たとえば、値 2dx ではページサイズが 45 バイトに設定されます。

MAX

ページサイズを動作環境で可能な最大値に設定します。4 バイト符号付き整数の最大値である $2^{31}-1$ (約 20 億バイト)以下の値になります。

詳細

ページサイズは、1 回の I/O 操作で 1 個のバッファに転送できるデータ量です。ページサイズは、データセットの永続的属性であり、データセットが処理されるときに使用されます。

ページサイズが大きいほど、ストレージメディアに対する必要な読み取りまたは書き込み回数を減らして、実行時間を短縮します。ただし、実行時間が改善するかわりにメモリ消費が増えます。

ページサイズを変更するには、DATA ステップを使用してデータセットをコピーし、新しいページを指定するか、SAS デフォルトを使用します。ページサイズを動作環境のデフォルト値にリセットするには、BUFSIZE=0 を使用します。

注: COPY プロシジャを使用してデータセットを別のエンジンで割り当てられた別のライブラリにコピーする場合、指定されたデータのページサイズは保持されません。

動作環境の情報

BUFSIZE=のデフォルト値は、動作環境に応じて決まり、順次アクセスを最適化するように設定されます。直接(ランダム)アクセスの処理速度を向上させるには、BUFSIZE=の値を変更する必要があります。直接アクセスのデフォルト設定と使用可能な設定については、動作環境向け SAS ドキュメントの BUFSIZE=データセットオプションを参照してください。

関連項目:

データセットオプション:

- “[BUFNO=データセットオプション](#)” (9 ページ)

システムオプション:

- “[BUFSIZE=システムオプション](#)” (SAS システムオプション: リファレンス)

CNTLLEV=データセットオプション

SAS データセットへの共有アクセスのレベルを指定します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	データセットコントロール
制限事項:	入力データセットについてのみ指定します。

構文

CNTLLEV=LIB | MEM | REC

構文の説明

LIB

同時アクセスがライブラリレベルでコントロールされることを指定します。ライブラリレベルのコントロールにより、同時アクセスでは、ライブラリへの更新プロセスが 1 つのみに制限されます。

MEM

同時アクセスが SAS データセット(メンバー)レベルでコントロールされることを指定します。メンバーレベルのコントロールにより、同時アクセスでは、SAS データセットへの更新または出力プロセスが 1 つのみに制限されます。データセットが更新または出力プロセス用に関われている場合、他の操作からそのデータセットにはアクセスできません。データセットが入力プロセス用に関われている場合、他の同時入力プロセスは許可されますが、更新または出力プロセスは許可されません。

REC

同時アクセスがオブザベーション(レコード)レベルでコントロールされることを指定します。レコードレベルのコントロールにより、同じ SAS データセットに対する複数の更新アクセスは許可されますが、同じオブザベーションの同時アクセスは拒否されます。

詳細

CNTLLEV=オプションでは、SAS データセットへの共有更新アクセスが拒否されるレベルを指定します。SAS データセットは、複数の SAS セッションによって、または 1 つのセッション内の複数のステートメント、ウィンドウまたはプロシジャによって開くことができます。SAS プロシジャのデフォルトでは、可能な最大同時アクセスレベルが許可されるとともに、データとデータ分析の一貫性が保証されます。そのため、通常は CNTLLEV=データセットオプションを使用しません。

このオプションを使用するのは次の場合です。

- SAS Component Language (SCL)、SAS/IML ソフトウェア、DATA ステッププログラミングなどのように、アプリケーションでデータへのアクセスをコントロールする場合
- データのメンバーレベルコントロールを提供しないインターフェースエンジンでデータにアクセスする場合

CNTLLEV=REC を使用し、SAS プロシジャでデータ分析の一貫性のためにメンバーレベルのコントロールが必要な場合、SAS ログに警告が出力されます。この警告は、分析中に他のプロセスによりデータが更新されると、不正確または予測できない結果が発生する可能性があることを示します。

例: 共有アクセスレベルの変更

次の例では、最初の SET ステートメントに、共有アクセスレベルでデフォルトのメンバーレベルコントロールよりレコードレベルコントロールを優先させるための CNTLLEV=データセットオプションが含まれます。2 つ目の SET ステートメントでは SAS データセットがデフォルトのメンバーレベルコントロールで開かれます。

```
set datalib.fuel (cntllev=rec) point=obsnum;
.
.
.
set datalib.fuel;
by area;
```

COMPRESS=データセットオプション

新しい出力 SAS データセットでのオブザベーションの圧縮方法を指定します。

- 該当要素:** DATA ステップおよび PROC ステップ
- カテゴリ:** データセットコントロール
- 制限事項:** 出力データセットにのみ使用します。
-

構文

COMPRESS=NO | YES | CHAR | BINARY

構文の説明

NO

新しく作成された SAS データセット内でオブザベーションは圧縮されないこと(固定長レコード)を指定します。

YES | CHAR

新しく作成された SAS データセット内でオブザベーションは SAS により RLE (Run Length Encoding)を使用して圧縮されること(可変長レコード)を指定します。RLE では、繰り返し連続する文字(空白を含む)を 2 バイトまたは 3 バイト表現に減らすことでオブザベーションを圧縮します。

別名 ON

BINARY

新しく作成された SAS データセット内でオブザベーションは SAS により RDC (Ross Data Compression)を使用して圧縮されること(可変長レコード)を指定します。RDC では、RLE (Run Length Encoding)とスライディングウィンドウ圧縮を組み合わせることでファイルを圧縮します。

注: この方法は、サイズが中～大(数百バイト以上)のバイナリデータ(数値変数)のブロックを圧縮する場合に非常に効果的です。この圧縮関数は一度に 1 つのレコードに対してのみ動作するため、効果的に圧縮するには数百バイト以上のレコード長が必要です。

詳細

ファイルの圧縮は、各オブザベーションの表現に必要なバイト数を減らすプロセスです。ファイル圧縮の利点として、ファイルのストレージ要件の削減、処理中のデータ読み取り/書き込みに必要な I/O 操作数の削減などがあります。ただし、圧縮ファイルの読み取りには(各オブザベーションの圧縮を解除するオーバーヘッドのために)より多くの CPU リソースが必要になります。状況によっては、圧縮後のファイルサイズが減らずに増えることがあります。

COMPRESS=データセットオプションは、個別のファイルを圧縮するために使用します。このオプションは、出力データセット、つまり DATA ステップの DATA ステートメント内または SAS プロシジャの OUT=オプション内に指定されたデータセットに対してのみ指定します。COMPRESS=データセットオプションは、SAS データファイル(メンバータイプ DATA)を作成する場合にのみ使用します。SAS ビューは、データが含まれていないため圧縮できません。

ファイルが圧縮された後、設定はファイルの永続的属性になります。つまり、設定を変更するには、ファイルを再作成する必要があります。そのため、ファイルを圧縮解除するには、圧縮ファイルをコピーする DATA ステップに COMPRESS=NO を指定します。

比較

COMPRESS=データセットオプションは、LIBNAME ステートメントの COMPRESS=オプションと COMPRESS=システムオプションよりも優先されます。

データセットオプション POINTOBS=YES(デフォルト)により、圧縮データセットを順次アクセスではなく、ランダムアクセス(オブザベーション番号を指定)で処理できるように定義されます。ランダムアクセスでは、オブザベーション番号を FSEDIT プロシジャや、SET および MODIFY ステートメントの POINT=オプションに指定できます。

圧縮ファイルを作成するとき、空き領域の追跡と再利用のために(データセットオプションまたはシステムオプションとして)REUSE=YES を指定することもできます。REUSE=YES を指定すると、新しいオブザベーションは、他のオブザベーションの更新または削除によって空いた領域に挿入されます。デフォルトの REUSE=NO が有効な場合、新しいオブザベーションは既存のファイルに追加されます。

POINTOBS=YES と REUSE=YES は、相互排他、つまり一緒に使用することはできません。REUSE=YES は、POINTOBS=YES よりも優先されます。そのため、REUSE=YES を設定すると、POINTOBS=NO が自動的に設定されます。

TAPE エンジンでは COMPRESS=データセットオプションがサポートされますが、COMPRESS=システムオプションはサポートされません。

XPORT エンジンでは圧縮はサポートされません。

関連項目:

データセットオプション:

- “POINTOBS=データセットオプション” (45 ページ)
- “REUSE=データセットオプション” (53 ページ)

ステートメント:

- “LIBNAME ステートメント” (SAS ステートメント: リファレンス)

システムオプション:

- “COMPRESS=システムオプション” (SAS システムオプション: リファレンス)
- “REUSE=システムオプション” (SAS システムオプション: リファレンス)

その他:

- “データファイルの圧縮” (SAS 言語リファレンス: 解説編 26 章)

DLDMGACTION=データセットオプション

SAS ライブラリ内の SAS データセットの損傷が検出されたときに実行するアクションを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

構文

DLDMGACTION=FAIL | ABORT | REPAIR | NOINDEX | PROMPT

構文の説明

FAIL

ただちにステップを停止し、エラーメッセージをログに発行します。これはバッチモードのデフォルトです。

ABORT

ステップを終了し、エラーメッセージをログに発行し、SAS セッションを終了します。

REPAIR

データファイルが切り捨てられていなければ、インデックスと一貫性制約を自動的に修復して再構築します。切り捨てられたデータセットを修復するには、PROC DATASETS で REPAIR ステートメントを使用します。警告メッセージがログに発行されます。これは対話型モードのデフォルトです。

NOINDEX

インデックスと一貫性制約なしでデータファイルを自動的に修復し、インデックスファイルを削除し、無効にしたインデックスと一貫性制約を反映してデータファイルを

更新して、データファイルを INPUT モードでのみ開くように制限します。無効になったインデックスと一貫性制約を修正または削除するには PROC DATASETS REBUILD ステートメントを実行するように指示する警告が SAS ログに書き込まれます。

参照項 “DLDMGACTION=データセットオプション” (15 ページ)
目

“無効化されたインデックスと一貫性制約の修復” (SAS 言語リファレンス: 解説編 36 章)

PROMPT

FAIL、ABORT、REPAIR、NOINDEX のいずれかのアクションを選択するように求めるダイアログボックスを表示します。

DROP=データセットオプション

入力データセットの場合は、指定された変数を処理対象から除外し、出力データセットの場合は、指定された変数をデータセットへの書き込み対象から除外します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: 変数制御

構文

DROP=*variable-1* <...*variable-n*>

構文の説明

variable-1 <...*variable-n*>

1 つ以上の変数名のリストを指定します。SAS で許可される形式の変数のリストを指定できます。

詳細

オプションが入力データセットに関連付けられている場合、変数は処理には使用できません。DROP=データセットオプションが出力データセットに関連付けられている場合、変数は出力データセットには書き込まれませんが、処理には使用できます。

比較

- DROP=データセットオプションと DROP ステートメントには次の違いがあります。
 - DATA ステップでは、DROP=データセットオプションは入力と出力の両方のデータセットに適用できます。DROP ステートメントは、出力データセットにのみ適用されます。
 - DATA ステップで複数の出力データセットを作成する場合、異なる変数を異なるデータセットに書き込むには DROP=データセットオプションを使用します。DROP ステートメントはすべての出力データセットに適用されます。
 - PROC ステップでは、DROP=データセットオプションのみを使用でき、DROP ステートメントは使用できません。

- KEEP=データセットオプションでは、処理対象にする変数のリストか、出力データセットへの書き込み対象とする変数のリストを指定します。

例

例 1: 入力から変数を除外する

この例では、変数 SALARY および GENDER は処理対象には含まれず、出力データセットにも書き込まれません。

```
data plan1 plan2;
  set payroll(drop=salary gender);
  if hired<'01jan98'd then output plan1;
  else output plan2;
run;
```

SALARY および GENDER は、DROP=により SET ステートメントで PAYROLL から読み取りができなくなるため、DATA ステップのどのロジックでも使用できません。

例 2: 変数をデータセットに書き込まずに処理する

この例では、SALARY および GENDER は PLAN2 には書き込まれませんが、PLAN1 には書き込まれます。

```
data plan1 plan2(drop=salary gender);
  set payroll;
  if hired<'01jan98'd then output plan1;
  else output plan2;
run;
```

関連項目:

データセットオプション:

- [“KEEP=データセットオプション” \(30 ページ\)](#)

ステートメント:

- [“DROP ステートメント” \(SAS ステートメント: リファレンス\)](#)

ENCRYPT=データセットオプション

出力 SAS データセットを暗号化するかどうかを指定します。

- 該当要素:** DATA ステップおよび PROC ステップ
- カテゴリ:** データセットコントロール
- 制限事項:** 出力データセットにのみ使用します。
-

構文

ENCRYPT=YES | NO

構文の説明

Yes

ファイルを暗号化します。この暗号化では、データセットに保存されたパスワードが使用されます。ENCRYPT=YES を指定した場合、少なくとも READ=または PW= データセットオプションを同時に指定する必要があります。この暗号化方法ではパスワードが使用されるため、暗号化されたデータセット上のパスワードは、データセットを再作成せずに変更することはできません。

注意:

ENCRYPT=YES を使用する場合はすべてのパスワードを記録してください。パスワードを忘れた場合、パスワードをリセットするには SAS のサポートが必要です。このプロセスには非常に多くの時間とリソースが必要になります。

NO

ファイルを暗号化しません。

詳細

ENCRYPT=YES を使用する場合、次のルールが適用されます。

- SAS ビューは、データが含まれていないため暗号化できません。
- 暗号化されたデータファイルをコピーするには、出力エンジンで暗号化がサポートされている必要があります。サポートされていない場合、データファイルはコピーされません。
- 暗号化されたファイルは、SAS 6.11 以降でのみ機能します。
- データファイルが暗号化されると、関連付けられたインデックスもすべて暗号化されます。
- 暗号化には、圧縮とほぼ同じ量の CPU リソースが必要です。
- PROC CPORT は、SAS 独自の暗号化データファイルに対しては使用できません。

例: ENCRYPT=YES オプションの使用

この例では、暗号化を使用して暗号化された SAS データセットを作成します。

```
data salary(encrypt=yes read=green);
input name $ yrsal bonuspct;
datalines;
Muriel 34567 3.2
Bjorn 74644 2.5
Freda 38755 4.1
Benny 29855 3.5
Agnetha 70998 4.1
;
```

このデータセットを使用するには、読み取りパスワードを指定します。

```
proc contents data=salary(read=green);
run;
```

関連項目:

データセットオプション:

- [“ALTER=データセットオプション” \(8 ページ\)](#)
- [“PW=データセットオプション” \(47 ページ\)](#)

- “READ=データセットオプション” (48 ページ)
- “WRITE=データセットオプション” (69 ページ)

その他:

- “SAS データファイルの暗号化” (SAS 言語リファレンス: 解説編 34 章)

EXTENDOBSCOUNTER=データセットオプション

新しい出力 SAS データファイル内の最大オブザベーションカウントを拡張するかどうかを指定します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	データセットコントロール
別名:	EOC=
デフォルト:デフォルト:	NO
制限事項:	出力データファイルにのみ使用します。 BASE エンジンでのみ使用します。

構文

EXTENDOBSCOUNTER=NO | YES

構文の説明

NO

新しく作成した SAS データファイル内での最大オブザベーションカウントを、動作環境の長整数型の大きさによって決めるように指定します。32 ビット長整数型を使用する動作環境では、最大数は $2^{31}-1$ 、つまり約 20 億(2,147,483,647)オブザベーションになります。64 ビット長整数型を使用する動作環境では、この数は $2^{63}-1$ 、つまり約 920 京オブザベーションになります。

Yes

新しく作成された SAS データファイルに、32 ビット長の制限を超えてオブザベーションを数える拡張ファイル形式を要求します。32 ビット長整数型で保存するオブザベーション数が決まる動作環境で作成された SAS データファイルの場合、そのデータファイルはカウンタに関しては 64 ビットファイルのように動作します。

制限事項 EXTENDOBSCOUNTER=YES を指定して作成された SAS データファイルには、SAS 9.3 よりも前のリリースとの互換性がありません。

EXTENDOBSCOUNTER=YES は、内部データ表現にオブザベーションカウントが 32 ビット長整数型として保存される出力 SAS データファイルにのみ指定します。EXTENDOBSCOUNTER=YES に適した動作環境と OUTREP=データ表現値の一覧表については、“EXTENDOBSCOUNTER=YES オプションを使用する場合” (SAS 言語リファレンス: 解説編 26 章)を参照してください。

関連項目:

“SAS データファイルのオブザベーションカウントの拡張” (SAS 言語リファレンス: 解説編 26 章)

FILECLOSE=データセットオプション

SAS データセットを閉じたときのテープの位置付けを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: その他

制限事項:

参照項目: UNIX 動作環境向けドキュメントの FILECLOSE=データセットオプション。

注意: このオプション値は、すべての動作環境で認識されるわけではありません。一部の動作環境では、追加の値を使用できます。テープに保存された SAS ライブラリの使用の詳細については、動作環境向け SAS ドキュメントの該当セクションを参照してください。

構文

FILECLOSE=DISP | LEAVE | REREAD | REWIND

構文の説明**DISP**

テープボリュームを、動作環境のコントロール言語で指定された配置に従って位置付けます。

LEAVE

テープを最後に処理したファイルの末尾に位置付けます。FILECLOSE=LEAVE は、SAS プログラムで同じファイルに繰り返しアクセスすることはなく、同じテープ上で後にある 1 つ以上の SAS ファイルにアクセスする場合に使用します。

REREAD

テープボリュームを最後に処理したファイルの先頭に位置付けます。FILECLOSE=REREAD は、SAS プログラムで同じテープ上にある同じ SAS データセットに複数回アクセスする場合に使用します。

REWIND

テープを最初まで巻き戻します。FILECLOSE=REWIND は、SAS プログラムで同じファイルに繰り返しアクセスすることはなく、同じテープ上で前にある 1 つ以上の SAS ファイルにアクセスする場合に使用します。

FIRSTOBS=データセットオプション

SAS データセット内で SAS が最初に処理するオブザベーションを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: オブザベーションコントロール

制限事項: 入力(読み取り)処理でのみ有効です。
PROC SQL ビューには使用できません。

構文

FIRSTOBS= *n* | *nK* | *nM* | *nG* | *hexX* | MIN | MAX

構文の説明

n | *nK* | *nM* | *nG*

最初に処理するオブザベーションの番号を、1(バイト)、1,024(キロバイト)、1,048,576(メガバイト)、1,073,741,824(ギガバイト)のいずれかの倍数で指定します。たとえば、値 8 では 8 番目のオブザベーション、値 3k では 3,072 が指定されます。

hexX

最初に処理するオブザベーションの番号を 16 進値で指定します。先頭が数値(0~9)、末尾が X の値を指定する必要があります。たとえば、値 2dx では 45 番目のオブザベーションが最初に処理するオブザベーションとして設定されます。

MIN

最初に処理するオブザベーションの番号を 1 に設定します。これがデフォルト設定です。

MAX

最初に処理するオブザベーションの番号を、データセットの最大オブザベーション数に設定します。8 バイト符号付き整数の最大値である $2^{63}-1$ (約 920 京オブザベーション)以下の値になります。

詳細

FIRSTOBS=データセットオプションは、既存の 1 つの SAS データセットに対してのみ有効です。現在の SAS セッション存続中のすべてのステップで有効にするには、FIRSTOBS=システムオプションを使用します。

FIRSTOBS=は入力(読み取り)処理でのみ有効です。FIRSTOBS=を指定しても、出力または更新処理には無効です。

WHERE 処理には FIRSTOBS=処理を適用できます。詳細については、“条件選択によるサブセットデータの処理”(SAS 言語リファレンス: 解説編 11 章)を参照してください。

比較

- FIRSTOBS=データセットオプションは、個々のデータセットの FIRSTOBS=システムオプションよりも優先されます。
- FIRSTOBS=データセットオプションでは処理の開始点を指定しますが、OBS=データセットオプションでは終了点を指定します。この 2 つのオプションは、多くの場合、処理するオブザベーションの範囲を定義するために使用されます。
- 外部ファイルを読み取る場合、INFILE ステートメントの FIRSTOBS=オプションで最初に読み取るレコードが指定されます。

例: 例

この PROC ステップでは、データセット STUDY をオブザベーション 20 から印刷します。

```
proc print data=study(firstobs=20);
run;
```

この SET ステートメントでは、FIRSTOBS=と OBS=の両方を使用して、データセット STUDY からオブザベーション 5~10 のみを読み取ります。データセット NEW には 6 個のオブザベーションが含まれます。

```
data new;
set study(firstobs=5 obs=10);
run;

proc print data=new;
run;
```

関連項目:

データセットオプション:

- “OBS=データセットオプション” (35 ページ)

ステートメント:

- “INFILE ステートメント” (SAS ステートメント: リファレンス)
- “WHERE ステートメント” (SAS ステートメント: リファレンス)

システムオプション:

- “FIRSTOBS=システムオプション” (SAS システムオプション: リファレンス)

GENMAX=データセットオプション

新しいデータセットの世代を要求し、既存のデータセットの世代数を変更し、バージョンの最大数を指定します。

- 該当要素:** DATA ステップおよび PROC ステップ
- カテゴリ:** データセットコントロール
- 制限事項:** 出力データセットにのみ使用します。
-

構文

GENMAX=*number-of-generations*

構文の説明

number-of-generations

データセットの世代を要求し、維持する最大バージョン数を指定します。0~1000 の値を指定できます。デフォルトの GENMAX=0 は、世代データセットを要求しないことを意味します。

詳細

GENMAX=は、新しいデータセットの世代を要求したり、既存のデータセットの世代数を変更したりする場合に使用します。データセットが初めて置き換えられるとき、SAS では置き換えられたバージョンが保持され、メンバー名に 4 文字のバージョン番号(#+3 桁の数値)が付加されます。たとえば、A という名前のデータセットの場合、履歴バージョンは A#001 になります。

データセットの世代が要求されると、メンバー名は 28 文字に制限されます(32 文字ではない)。末尾の 4 文字は、付加されるバージョン番号用に予約されます。GENMAX=データセットオプションを 0 に設定すると、メンバー名を最大 32 文字にすることができます。

既存のデータセットの世代数を減らすと、新しい制限を超えるバージョンは SAS により古い方から削除されます。

例

例 1: データセット作成時の世代の要求

この例では、新しいデータセットの世代を要求する方法を示します。DATA ステップで WORK.A という名前のデータセットが作成されます。このデータセットは最大 10 世代を持つことができます(1 個の最新バージョンと 9 個の履歴バージョン)。

```
data a (genmax=10);
x=1;
output;
run;
```

例 2: 既存のデータセットの世代数の変更

この例は、データセット MYLIB.A の世代数を 4 に変更する方法を示します。

```
proc datasets lib=mylib;
modify a (genmax=4);
run;
```

関連項目:

データセットオプション:

- “GENNUM=データセットオプション” (23 ページ)

その他:

- “世代データセットについて” (SAS 言語リファレンス: 解説編 26 章)

GENNUM=データセットオプション

SAS データセットの特定の世代を指定します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	データセットコントロール
制限事項:	入力データセットにのみ使用します。

構文

GENNUM=*integer*

構文の説明

integer

世代グループ内の特定のバージョンを参照する番号です。正の数を指定すると、データセット名に付加されている特定の世代番号が絶対参照されます。負の数を指定すると、基本バージョンに対する履歴バージョンが新しい順に相対参照されます。通常、値 0 が現在の(基本)バージョンを示します。

DATASETS プロシジャでは、さまざまなステートメントで GENNUM=を指定して追加機能を使用できます。

- DATASETS および DELETE ステートメントの場合、GENNUM=では追加の値 ALL、HIST および REVERT がサポートされます。
- CHANGE ステートメントの場合、GENNUM=では追加の値 ALL がサポートされます。
- CHANGE ステートメントの場合、GENNUM=0 を指定すると基本バージョンだけでなくすべてのバージョンを示します。

詳細

GENMAX=データセットオプションを使用してデータセットの世代を要求した後、特定のバージョンを要求するには GENNUM=を使用します。たとえば、GENNUM=3 を指定すると履歴バージョン#003 を示し、GENNUM=-1 を指定すると最も新しい履歴バージョンを示します。

置き換えが 999 回行われると、最も新しいバージョンは#999 になります。置き換えが 1,000 回行われると、最も新しいバージョン番号は#000 に戻ります。そのため、履歴バージョン#000 が必要な場合は、GENNUM=1000 を指定します。

絶対参照と相対参照のどちらも、特定のバージョンを参照します。相対参照では、削除されたバージョンがスキップされません。そのため、削除されたバージョンが 1 つ以上含まれる世代グループを操作する場合、相対参照を使用して参照したバージョンが削除されているとエラーが発生します。たとえば、基本バージョン AIR と 3 つの履歴バージョン(AIR#001、AIR#002、AIR#003)があり、AIR#002 を削除した場合、次のステートメントでは AIR#002 が存在しないためにエラーが返されます。SAS では、この指定が AIR#003 を意味するとは想定されません。

```
proc print data=air (gennum= -2);
run;
```

例

例 1: 絶対参照を使用してバージョンを要求する

この例では、絶対参照を使用してデータセット A の履歴バージョン#003 を印刷します。

```
proc print data=a(gennum=3);
run;
```

例 2: 相対参照を使用してバージョンを要求する

次の PRINT プロシジャでは、基本バージョンから 3 バージョン前のデータセットを印刷します。

```
proc print data=a(gennum=-3);
run;
```

関連項目:**データセットオプション:**

- “GENMAX=データセットオプション” (22 ページ)

その他:

- “世代データセットについて” (SAS 言語リファレンス: 解説編 26 章)
- “DATASETS プロシジャ” (Base SAS プロシジャガイド)

IDXNAME=データセットオプション

WHERE 式の条件の照合に特定のインデックスを使用するように SAS に指示します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	SAS インデックス使用のユーザーコントロール
制限事項:	入力データセットのみで使用 IDXWHERE=データセットオプションと相互排他的

構文

IDXNAME=*index-name*

構文の説明*index-name*

SAS データセットの単一インデックスまたは複合インデックスの名前(32 文字以内)を指定します。指定したインデックスが最適なものであるかどうか、順次検索の方がリソースをより効率的に利用できるかどうかについて SAS は確認しません。

操作 この指定はデータセットの永続的な属性ではなく、現在の使用に対してのみ有効です。

ヒント IDXNAME=の使用を SAS ログに表示するように要求するには、システムオプション MSGLEVEL=I を指定します。

詳細

デフォルトでは、インデックス付き SAS データセットの WHERE 式の条件を満たすため、SAS は WHERE 式の最適化に使用できる 0 個以上の候補インデックスを識別します。SAS は候補インデックスのリストから処理速度が最適なインデックスを決定するか、データを順次に確認した方がより効率的な場合はすべてのインデックスを拒否します。

SAS が選択するインデックスは常に最適であるとは限らないため、IDXNAME=データセットオプションを指定することで、いずれかの候補インデックスを使用するように SAS に指示できます。SAS によって候補インデックスとして識別されないインデックスを指定した場合、IDXNAME=では要求が処理されません。つまり、IDXNAME=には結果が無効となるインデックスを指定できません。

比較

IDXWHERE=では、インデックスを使用するかどうかの SAS の決定を変更できます。

例: 例

この例では、IDXNAME=データセットオプションを使用して、WHERE 式の最適化に特定のインデックスを使用するように SAS に指示します。SAS はデータセットの順次検索の方がリソースをより効率的に利用できる可能性を無視し、指定したインデックスが最適なものであるかどうかを確認しません (EMPNUM インデックスは NOMISS オプションを使用して作成されていません)。

```
data mydata.empnew;
set mydata.employee (idxname=empnum);
where empnum < 2000;
run;
```

関連項目:

データセットオプション:

- “IDXWHERE=データセットオプション” (26 ページ)

その他:

- “WHERE 式処理に対するインデックスの使用” (SAS 言語リファレンス: 解説編 26 章)

IDXWHERE=データセットオプション

SAS で WHERE 式の条件の照合にインデックス検索を使用するか、順次検索を使用するかを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: SAS インデックス使用のユーザーコントロール

制限事項: 入力データセットにのみ使用します。

IDXNAME=データセットオプションと相互排他的

構文

IDXWHERE=YES | NO

構文の説明

Yes

WHERE 式の最適化に最適なインデックスを選択し、データセットの順次検索の方がリソースをより効率的に利用できる可能性を無視するように SAS に指示します。

NO

すべてのインデックスを無視し、データセットの順次検索で WHERE 式の条件を満たすように SAS に指示します。

注: IDXWHERE=を使用して、BY ステートメントの処理に対するインデックスの使用を変更することはできません。

詳細

デフォルトでは、インデックス付き SAS データセットの WHERE 式の条件を満たすため、SAS はインデックスを使用するか、データセットを順次に読み取るかを決定します。ソフトウェアによって相対的な効率性が推定され、より効率的な方法が選択されま

す。
ソフトウェアの決定は、最適な結果にならない可能性もある一般的なルールに基づくため、IDXWHERE=データセットオプションの指定によるこの決定の変更が必要な場合があります。つまり、IDXWHERE=データセットオプションを指定することによって、処理方法を決定できます。

注: この指定はデータセットの永続的な属性ではなく、現在の使用に対してのみ有効です。

注: システムオプション MSGLEVEL=I を指定すると、設定がインデックス処理に影響する場合は IDXWHERE=の使用を SAS ログに表示するように要求できます。

比較

IDXNAME=では、特定のインデックスを使用するように SAS に指示できます。

例

例 1: インデックスの使用を指定する

この例では、IDXWHERE=データセットオプションを使用して、WHERE 式の最適化に最適なインデックスを決定するように SAS に指示します。SAS は、データセットの順次検索の方がリソースをより効率的に利用できる可能性を無視します。

```
data mydata.empnew;
set mydata.employee (idxwhere=yes);
where empnum < 2000;
```

例 2: インデックスを使用しないように指定する

この例では、IDXWHERE=データセットオプションを使用して、すべてのインデックスを無視し、データセットの順次検索で WHERE 式の条件を満たすように SAS に指示します。

```
data mydata.empnew;
set mydata.employee (idxwhere=no);
where empnum < 2000;
```

関連項目:

データセットオプション:

- [“IDXNAME=データセットオプション” \(25 ページ\)](#)

その他:

- [“SAS インデックスについて” \(SAS 言語リファレンス: 解説編 26 章\)](#)
- [WHERE 式の処理](#)

INDEX=データセットオプション

新しい出力 SAS データセットのインデックスを定義します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	データセットコントロール
制限事項:	出力データセットにのみ使用します。

構文

`INDEX=(index-specification-1 ...<index-specification-n>)`

構文の説明

index-specification

作成する単一インデックスまたは複合インデックスの名前および説明を指定します。*Index-specification* は次のような形式になります。

`index= (variable </UNIQUE> </NOMISS>)`

index

インデックスを形成する変数名または複合インデックス用に選択する名前です。

variable

複合インデックスの作成に使用する変数のリストです。

UNIQUE

主要変数の値は一意である必要があることを指定します。新しいデータセットに UNIQUE を指定し、複数のオブザベーションのインデックス変数の値が同じ場合、インデックスは作成されません。UNIQUE オプションの前にスラッシュ(/)を付ける必要があります。

NOMISS

欠損値を含むすべてのオブザベーションをインデックスから除外します。欠損値を含むオブザベーションは引き続きデータセットから読み込まれますが、インデックスは使用されません。NOMISS オプションの前にスラッシュ(/)を付ける必要があります。

例

例 1: 単一インデックスの定義

次の INDEX=データセットオプションでは、SSN 変数の単一インデックスを定義します。

```
data new(index=(ssn));
```

例 2: 複合インデックスの定義

次の INDEX=データセットオプションでは、CITY 変数および STATE 変数を使用する CITYST という名前の複合インデックスを定義します。

```
data new(index=(cityst=(city state)));
```


例 3: 単一インデックスおよび複合インデックスの定義

次の INDEX=データセットオプションでは、SSN の単一インデックスおよび CITY と STATE の複合インデックスを定義します。

```
data new(index=(ssn cityst=(city state)));
```

例 4: UNIQUE オプションを使用した単一インデックスの定義

次の INDEX=データセットオプションでは、一意の値の SSN 変数の単一インデックスを定義します。

```
data new(index=(ssn /unique));
```

例 5: NOMISS オプションを使用した単一インデックスの定義

次の INDEX=データセットオプションでは、欠損値を含むすべてのオブザベーションをインデックスから除外した SSN 変数の単一インデックスを定義します。

```
data new(index=(ssn /nomiss));
```

例 6: UNIQUE オプションおよび NOMISS オプションを使用して複数のインデックスを定義する

次の INDEX=データセットオプションでは、SSN 変数の単一インデックスおよび CITY と STATE の複合インデックスを定義します。各変数に UNIQUE オプションおよび NOMISS オプションが設定されている必要があります。

```
data new(index=(ssn /unique/nomiss cityst=(city state)/unique/nomiss));
```

関連項目:

- “INDEX CREATE ステートメント” (Base SAS プロシジャガイド 15 章)
- “CREATE INDEX ステートメント” (SAS SQL プロシジャユーザーガイド 7 章)
- “SAS インデックスについて” (SAS 言語リファレンス: 解説編 26 章)

IN=データセットオプション

データセットのデータが現在のオブザベーションに関与しているかどうかを示すブール値変数を作成します。

該当要素: DATA ステップ

カテゴリ: オブザベーションコントロール

制限事項: SET、MERGE、MODIFY、UPDATE ステートメントのみとともに使用します。

構文

```
IN=variable
```

構文の説明**変数**

入力データセットのデータが現在のオブザベーションに関与しているかどうかを示す値を含む、新しい変数の名前を指定します。DATA ステップ内での変数の値は、データセットが現在のオブザベーションに関与している場合は 1、それ以外の場合は 0 です。

詳細

IN=データセットオプションは、SET、MERGE、MODIFY、UPDATE ステートメントのみで SAS データセット名の後にかっこで囲んで指定します。IN=変数の値は、DATA ステップ内のプログラムステートメントで使用できます。新しく作成される SAS データセットでは、新しい変数を割り当てない限り変数は含まれません。

BY-グループ処理で IN=を使用し、現在の BY グループでデータセットがオブザベーションに参与する場合、IN=の値は 1 です。この値は BY グループの処理中はそのまま保持され、プログラミングロジックによってリセットされません。

例

この例では、国際飛行便を示す新しい変数、OVERSEAS を IN=で作成します。変数 I の値は、オブザベーションが NONUSA データセットから読み込まれている場合は 1 になります。それ以外の場合、この値は 0 になります。IF-THEN ステートメントで I の値を確認し、データセット NONUSA のデータが現在のオブザベーションに参与しているかどうかを判断します。I=1 の場合、変数 OVERSEAS には値としてアスタリスク(*)が渡されます。

```
data allflts;
  set usa nonusa(in=i);
  by fltnum;
  if i then overseas='*';
run;
```

関連項目:

ステートメント:

- “BY ステートメント” (SAS ステートメント: リファレンス)
- “MERGE ステートメント” (SAS ステートメント: リファレンス)
- “MODIFY ステートメント” (SAS ステートメント: リファレンス)
- “SET ステートメント” (SAS ステートメント: リファレンス)
- “UPDATE ステートメント” (SAS ステートメント: リファレンス)

その他:

- BY-グループ処理

KEEP=データセットオプション

入力データセットの場合は、処理する変数を指定し、出力データセットの場合は、データセットに書き込む変数を指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: 変数制御

構文

KEEP=*variable-1* <...*variable-n*>

構文の説明

variable-1 <..*variable-n*>

1 つ以上の変数名のリストを指定します。SAS で許可される形式の変数のリストを指定できます。

詳細

KEEP=データセットオプションが入力データセットに関連付けられている場合、KEEP=データセットオプションの後に示された変数のみを処理に使用できます。KEEP=データセットオプションが出力データセットに関連付けられている場合、このオプションの後に示された変数のみが出力データセットに書き込まれます。すべての変数を処理に使用できます。

比較

- KEEP=データセットオプションと KEEP ステートメントには次の違いがあります。
 - DATA ステップでは、KEEP=データセットオプションは入力と出力の両方のデータセットに適用できます。KEEP ステートメントは、出力データセットにのみ適用されます。
 - DATA ステップで複数の出力データセットを作成する場合、異なる変数を異なるデータセットに書き込むには KEEP=データセットオプションを使用します。KEEP ステートメントは、すべての出力データセットに適用されます。
 - PROC ステップでは、KEEP=データセットオプションのみを使用でき、KEEP ステートメントは使用できません。
- DROP=データセットオプションでは、処理中に除外する変数または出力データセットから除外する変数を指定します。

例

この例では、IDNUM および SALARY のみが PAYROLL から読み込まれ、この 2 つのみが処理で使用できる PAYROLL 内の変数です。

```
data bonus; set payroll(keep=idnum salary); bonus=salary*1.1; run;
```

関連項目:

データセットオプション:

- “DROP=データセットオプション” (16 ページ)

ステートメント:

- “KEEP ステートメント” (SAS ステートメント: リファレンス)

LABEL=データセットオプション

SAS データセットのラベルを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

構文

```
LABEL='label'
```

構文の説明

'label'

256 文字までのテキスト文字列を指定します。ラベルテキストに一重引用符が含まれる場合、ラベルを二重引用符で囲みます。または、ラベルテキスト内で 2 つの一重引用符を使用し、文字列を一重引用符で囲むこともできます。データセットからラベルを削除するには、引用符で囲んだ空白に相当するラベルを割り当てます。

詳細

LABEL=オプションは、入力データセットと出力データセットの両方で使用できます。入力データセットで LABEL=を使用する場合、ファイルに割り当てられるラベルは DATA ステップまたは PROC ステップの期間中のみ有効です。出力データセットで LABEL=を指定する場合、ラベルは永続的にそのファイルのラベルになります。ファイルを出力するには CONTENTS または DATASETS プロシジャを使用し、変更するには PROC DATASETS を使用します。

データセットに割り当てられたラベルは、APPEND プロシジャまたは MODIFY ステートメントなどを使用してデータセットを更新する場合はそのデータセットに関連付けられたままになります。ただし、以前にラベルが割り当てられているデータセットを使用して DATA ステップで新しいデータセットを作成する場合、ラベルは失われます。たとえば、データセット ONE に以前に割り当てられたラベルは、この DATA ステップで新しい出力データセット ONE を作成したときに失われます。

```
data one;
set one;
run;
```

比較

- LABEL=データセットオプションでは、データセットのみにラベルを指定できます。データセット内の変数のラベルは、LABEL ステートメントを使用して指定できません。
- ATTRIB ステートメントの LABEL=オプションでも、変数にラベルを割り当てることができます。

例

データセットにラベルを割り当てる例を次に示します。

```
data w2(label='1976 W2 Info, Hourly');
data new(label='Peter''s List');
data new(label="Hillside's Daily Account");
data sales(label='Sales For May(NE)');
```

関連項目:

ステートメント:

- “ATTRIB ステートメント” (SAS ステートメント: リファレンス)
- “LABEL ステートメント” (SAS ステートメント: リファレンス)

- “MODIFY ステートメント” (SAS ステートメント: リファレンス)

プロシジャ:

- “CONTENTS プロシジャ” (Base SAS プロシジャガイド)
- “DATASETS プロシジャ” (Base SAS プロシジャガイド)

OBSBUF=データセットオプション

DATA ステップビューを処理する表示バッファのサイズを決定します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	データセットコントロール
制限事項:	DATA ステップビューでのみ有効です。

構文

OBSBUF=*n*

構文の説明

n

一度に表示バッファに読み込まれるオブザベーション数を指定します。

デフォルト デフォルトの表示バッファには 32K バイトのメモリが割り当てられます。つまり、一度に表示バッファに読み込むことができるデフォルトのオブザベーション数は、オブザベーションの長さによって異なります。そのため、デフォルトは 32K バイトに収まるオブザベーション数です。オブザベーションの長さが 32K を超える場合、一度にバッファに読み込むことができるオブザベーションは 1 つのみです。

ヒント オブザベーションの長さ(バイト数でのサイズ)を確認するには、DATA ステップビューの PROC CONTENTS を使用します。

注意 OBSBUF=オプションの最大値は、使用可能なメモリ量によって異なります。指定した値が大きすぎて表示バッファのメモリ割り当てに失敗した場合、メモリ不足エラーが発生します。使用可能な実メモリ量よりも大きい値を指定し、SAS が仮想メモリを使用して割り当てを実行できる動作環境の場合、過剰なページングによって処理速度が低下する可能性があります。

詳細

OBSBUF=データセットオプションでは、一度に表示バッファに読み込むことができるオブザベーション数を指定します。表示バッファは、DATA ステップビューから生成された出力オブザベーションを保持するために割り当てられたメモリのセグメントです。バッファのサイズによって、一度にメモリ内に保持できるデータ量が決定されます。OBSBUF=では、DATA ステップビューからのデータの読み込み処理速度を調整できます。

表示バッファは、DATA ステップビューを開く要求(SAS プロシジャなど)と DATA ステップビュー自体の間で共有されます。2 つのコンピュータタスクは、データの要求とデータの生成や返送間で次のように調整されます。

1. 要求タスク(PRINT プロシジャなど)がデータを要求すると、DATA ステップビューを実行してオブザベーションを生成するために、要求タスクから表示タスクへのタスクの切り替えが発生します。DATA ステップビューによって、表示バッファに可能な限り多くのオブザベーションが挿入されます。
2. 表示バッファがいっぱいになると、要求されたデータを返すために表示タスクから要求タスクへのタスクの切り替えが発生します。オブザベーションは表示バッファから解除されます。

表示バッファのサイズによって、保持できる生成されたオブザベーション数が決定されます。次に、生成されたオブザベーション数によって、コンピュータが要求タスクと表示タスク間を切り替える必要がある回数決定されます。たとえば、OBSBUF=1 では各オブザベーションごとにタスクの切り替えが発生し、OBSBUF=10 では一度に 10 個のオブザベーションが表示バッファに読み込まれます。表示バッファを大きくすると、DATA ステップビューの処理に必要なタスクの切り替え数は少なくなり、実行時間が短縮されます。

効率を高めるには、最初にデフォルトのバッファサイズに収まるオブザベーション数を確認してから、より多くの生成されたオブザベーションを保持できるように表示バッファを設定します。

注: OBSBUF=を使用し、タスクの切り替え数を減らすことで処理効率を高めることができます。ただし、表示バッファサイズを大きくすると、バッファがいっぱいになるまでにかかる時間が長くなります。これにより、要求されたデータを返すための表示タスクから要求タスクへのタスクの切り替えで遅延が発生します。この遅延は、対話型アプリケーションでより顕著です。たとえば、**テーブルの表示ウィンドウ**を使用する場合、表示バッファを大きくすると、要求したオブザベーションの表示にかかる時間が長くなります。1 個のオブザベーションのみが**テーブルの表示**に返される場合でも、その前に表示バッファがいっぱいになります。そのため、非常に大きい表示バッファサイズを設定する前に、DATA ステップビューの処理に使用するアプリケーションの種類、および使用可能なメモリ量を考慮してください。

例

この例では、オブザベーションの長さは 10K です。そのため、デフォルトの表示バッファサイズ(32K)では、一度に 3 個のオブザベーションが表示バッファに読み込まれます。デフォルトの表示バッファサイズでは、3 個のオブザベーションが生成されるたびにコンピュータがタスクを切り替える必要があるため、実行時間が長くなります。

処理速度を向上させるには、OBSBUF=データセットオプションを 100 に設定します。この設定では、一度に 100 個のオブザベーションが表示バッファに読み込まれます。また、PRINT プロシジャで DATA ステップビューを処理するためのタスクの切り替え数が削減されます。

```
data testview / view=testview;
... more SAS statements ...
run;
proc print data=testview (obsbuf=100);
run;
```

関連項目:

データセットオプション:

- “SPILL=データセットオプション” (57 ページ)

OBS=データセットオプション

SAS が処理するデータセット内の最後のオブザベーションを指定します。

該当要素:	DATA ステップおよび PROC ステップ
カテゴリ:	オブザベーションコントロール
デフォルト:デフォルト:	MAX
制限事項:	入力データセットのみで使用 PROC SQL ビューでは使用できません。

構文

OBS= *n* | *nK* | *nM* | *nG* | *nT* | *hexX* | MIN | MAX

構文の説明

n | *nK* | *nM* | *nG* | *nT*

オブザベーションの処理を停止するタイミングを示す数を整数 *n* で指定します。いずれかの文字表記を使用すると、整数が特定の値で乗算されます。具体的には、指定表記が K (キロ) の場合は 1,024、M (メガ) の場合は 1,048,576、G (ギガ) の場合は 1,073,741,824、T (テラ) の場合は 1,099,511,627,776 で整数が乗算されます。たとえば、値 20 では 20 個のオブザベーション、値 3m では 3,145,728 個のオブザベーションが指定されます。

hexX

処理を停止するタイミングを示す数を 16 進値で指定します。先頭が数値(0~9)、末尾が X の値を指定する必要があります。たとえば、10 進値の 248 に相当する 16 進値 F8 を指定するには、0F8x と指定する必要があります。値 2dx では、10 進値での 45 が指定されます。

MIN

処理を停止するタイミングを示す数を 0 に設定します。別のデータセットの構造を含み、オブザベーションを含まない空のデータセットを作成する場合に OBS=0 を使用します。

操作 OBS=0 で NOREPLACE オプションが有効になっている場合、SAS は特定の操作を実行できません。SAS は、オブザベーションを使用せずにプログラムの各 DATA ステップおよび PROC ステップを実際に行います。たとえば、ライブラリまたは SAS データセットを処理する、CONTENTS および DATASETS などのプロシジャを実行します。

MAX

処理を停止するタイミングを示す数をデータセット内の最大オブザベーション数に設定します。8 バイト符号付き整数の最大値である $2^{63}-1$ (約 920 京) 以下の値になります。これがデフォルト設定です。

詳細

OBS=では、オブザベーションの処理を停止するタイミングを SAS に指示します。処理を停止するタイミングを判断するため、SAS は計算式で OBS=の値を使用します。この計算式には OBS=の値と FIRSTOBS=の値が含まれます。式は次のとおりです。

```
(obs - firstobs) + 1 = results
```

たとえば、OBS=10 で FIRSTOBS=1(FIRSTOBS=のデフォルト値)の場合、結果は $(10 - 1) + 1 = 10$ で 10 個のオブザベーションになります。OBS=10 で FIRSTOBS=2 の場合、結果は $(10 - 2) + 1 = 9$ で 9 個のオブザベーションになります。OBS= は、既存の SAS データセットが読み込まれている場合にのみ有効です。

比較

- OBS=データセットオプションは、個々のデータセットの OBS=システムオプションよりも優先されます。
- OBS=データセットオプションでは処理の終点を指定しますが、FIRSTOBS=データセットオプションでは開始点を指定します。この 2 つのオプションは、多くの場合、処理するオブザベーションの範囲を定義するために使用されます。
- OBS=データセットオプションでは、SAS データセットからオブザベーションを選択できます。INFILE ステートメントで OBS=オプションを使用することで、外部データファイルから読み込まれるオブザベーションを選択できます。

例

例 1: OBS=を使用したオブザベーションの処理を停止するタイミングを指定する

この例では、OBS=を使用してオブザベーションの処理を停止するタイミングを SAS に指示した結果を示します。SAS データセットを作成し、FIRSTOBS=2 および OBS=12 を設定した PRINT プロシジャを実行します。この結果は $(12 - 2) + 1 = 11$ で 11 個のオブザベーションになります。この場合の出力はオブザベーション 2 で開始してオブザベーション 12 で終了するため、OBS=の結果は SAS が最後に処理するオブザベーション番号であるように見えます。この状況は単なる偶然に過ぎません。

```
data Ages;
input Name $ Age;
datalines;
Miguel 53
Brad 27
Willie 69
Marc 50
Sylvia 40
Arun 25
Gary 40
Becky 51
Alma 39
Tom 62
Kris 66
Paul 60
Randy 43
Barbara 52
Virginia 72
;
proc print data=Ages (firstobs=2 obs=12);
run;
```


画面 2.1 OBS=およびFIRSTOBS=を使用した PROC PRINT の出力

The SAS System

Obs	Name	Age
2	Brad	27
3	Willie	69
4	Marc	50
5	Sylvia	40
6	Arun	25
7	Gary	40
8	Becky	51
9	Alma	39
10	Tom	62
11	Kris	66
12	Paul	60

例 2: WHERE 処理で OBS=を使用する

この例では、WHERE 処理とともに OBS=を使用した結果を示します。例 1 で作成された、15 個のオブザベーションを含むデータセットを使用します。

最初に、WHERE ステートメントを含む PRINT プロシジャを次に示します。データのサブセットの結果は 12 個のオブザベーションになります。

```
proc print data=Ages;  
where Age LT 65;  
run;
```

画面 2.2 WHERE ステートメントを使用した PROC PRINT の出力


The SAS System

Obs	Name	Age
1	Miguel	53
2	Brad	27
4	Marc	50
5	Sylvia	40
6	Arun	25
7	Gary	40
8	Becky	51
9	Alma	39
10	Tom	62
12	Paul	60
13	Randy	43
14	Barbara	52

WHERE ステートメントおよび OBS=10 を使用した PRINT プロシジャを実行すると、結果は $(10 - 1) + 1 = 10$ で 10 個のオブザベーションになります。SAS は最初に WHERE 処理でデータをサブセット化し、そのサブセットに OBS= を適用します。

```
proc print data=Ages (obs=10);  
  where Age LT 65;  
run;
```

画面 2.3 WHERE ステートメントおよび OBS=を使用した PROC PRINT の出力



The screenshot shows a window titled "The SAS System" containing a table with three columns: "Obs", "Name", and "Age". The table contains 12 rows of data, with the "Obs" column values being 1, 2, 4, 5, 6, 7, 8, 9, 10, and 12. The "Name" column contains names: Miguel, Brad, Marc, Sylvia, Arun, Gary, Becky, Alma, Tom, and Paul. The "Age" column contains ages: 53, 27, 50, 40, 25, 40, 51, 39, 62, and 60. The table is displayed in a light blue grid format.

Obs	Name	Age
1	Miguel	53
2	Brad	27
4	Marc	50
5	Sylvia	40
6	Arun	25
7	Gary	40
8	Becky	51
9	Alma	39
10	Tom	62
12	Paul	60

出力が 10 個のオブザベーションで構成されており、オブザベーション番号 12 で終了するため、OBS=の結果が処理するオブザベーション数であるように見えます。ただし、この結果は単なる偶然に過ぎません。FIRSTOBS=2 および OBS=10 をサブセットに適用した場合、結果は $(10 - 2) + 1 = 9$ で 9 個のオブザベーションになります。この場合の OBS=は最後のオブザベーション番号でも処理するオブザベーション数でもありません。値は処理を停止するタイミングを判別する計算式で使用されます。

```
proc print data=Ages (firstobs=2 obs=10);  
  where Age LT 65;  
run;
```

画面 2.4 WHERE ステートメント、OBS=およびFIRSTOBS=を使用した PROC PRINT の出力



Obs	Name	Age
2	Brad	27
4	Marc	50
5	Sylvia	40
6	Arun	25
7	Gary	40
8	Becky	51
9	Alma	39
10	Tom	62
12	Paul	60

この例では、削除されたオブザベーションがあるデータセットに OBS=を使用した結果を示します。例 1 で作成されたデータセットからオブザベーション 6 が削除されたデータセットを使用します。

最初に、変更されたファイルの PROC PRINT の出力を次に示します。

```
proc print data=Ages;  
run;
```

画面 2.5 オブザベーション 6 が削除されたことを示す PROC PRINT の出力


The SAS System

Obs	Name	Age
1	Miguel	53
2	Brad	27
3	Willie	69
4	Marc	50
5	Sylvia	40
7	Gary	40
8	Becky	51
9	Alma	39
10	Tom	62
11	Kris	66
12	Paul	60
13	Randy	43
14	Barbara	52
15	Virginia	72

OBS=12 を使用した PRINT プロシジャを実行すると、結果は $(12 - 1) + 1 = 12$ で 12 個のオブザベーションになります。

```
proc print data=Ages (obs=12);  
run;
```

画面 2.6 OBS=を使用した PROC PRINT の出力




Obs	Name	Age
1	Miguel	53
2	Brad	27
3	Willie	69
4	Marc	50
5	Sylvia	40
7	Gary	40
8	Becky	51
9	Alma	39
10	Tom	62
11	Kris	66
12	Paul	60
13	Randy	43

例 3: オブザベーションが削除された場合に OBS=を使用する

出力が 12 個のオブザベーションで構成されており、オブザベーション番号 13 で終了するため、OBS=の結果が処理するオブザベーション数であるように見えます。ただし、FIRSTOBS=2 および OBS=12 を適用した場合、結果は $(12 - 2) + 1 = 11$ で 11 個のオブザベーションになります。この場合の OBS=は最後のオブザベーション番号でも処理するオブザベーション数でもありません。値は処理を停止するタイミングを判別する計算式で使用されます。

```
proc print data=Ages (firstobs=2 obs=12);  
run;
```

画面 2.7 OBS=および FIRSTOBS=を使用した PROC PRINT の出力



The SAS System

Obs	Name	Age
2	Brad	27
3	Willie	69
4	Marc	50
5	Sylvia	40
7	Gary	40
8	Becky	51
9	Alma	39
10	Tom	62
11	Kris	66
12	Paul	60
13	Randy	43

関連項目:**データセットオプション:**

- “FIRSTOBS=データセットオプション” (20 ページ)

ステートメント:

- “INFILE ステートメント” (SAS ステートメント: リファレンス)
- “WHERE ステートメント” (SAS ステートメント: リファレンス)
- SAS ステートメント: リファレンス

システムオプション:

- “OBS=システムオプション” (SAS システムオプション: リファレンス)

OBS=および WHERE 処理の使用方法については、次を参照してください。

- “条件選択によるサブセットデータの処理” (SAS 言語リファレンス: 解説編 11 章)

OUTREP=データセットオプション

出力 SAS データセットのデータ表記を指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

参照項目: z/OS 版 SAS の OUTREP=データセットオプション

構文

OUTREP=*format*

構文の説明

format

特定の動作環境でデータを保存するときの形式のデータ表記を指定します。動作環境が異なると、使用される標準や規則も異なります。たとえば、浮動小数点数を保存する場合は IEEE や IBM メインフレーム、文字エンコーディングの場合は ASCII または EBCDIC、メモリ内のバイトの順序付けではビッグエンディアンまたはリトルエンディアン、ワード配置では 4 バイト境界または 8 バイト境界、整数のデータ型の長さでは 16 ビット、32 ビットまたは 64 ビットが使用され、倍精度の場合はバイト交換される場合とされない場合があります。

デフォルトでは、SAS を実行している CPU のデータ表記を使用して新しい SAS データセットが作成されます。OUTREP=オプションを指定することで、異なるデータ表記を使用して SAS データセットを作成できます。たとえば、UNIX 環境で、Windows データ表記を使用する SAS データセットを作成できます。互換性およびデータ表記の詳細については、“クロス環境データアクセス(CEDA)を用いたデータ処理” (SAS 言語リファレンス: 解説編 32 章)を参照してください。

OUTREP=の値については、次の表を参照してください。

表 2.1 OUTREP=オプションのデータ表記の値

OUTREP=の値	別名*	環境
ALPHA_TRU64	ALPHA_OSF	Tru64 UNIX
ALPHA_VMS_32	ALPHA_VMS	OpenVMS Alpha
ALPHA_VMS_64		OpenVMS Alpha
HP_IA64	HP_ITANIUM	HP-UX (Itanium プロセッサファミリアーキテクチャ)
HP_UX_32	HP_UX	HP-UX (PA-RISC)
HP_UX_64		HP-UX (PA-RISC)、64 ビット
INTEL_ABI		ABI (Intel アーキテクチャ)
LINUX_32	LINUX	Linux (Intel アーキテクチャ)
LINUX_IA64		Linux (Itanium ベースシステム)

OUTREP=の値	別名*	環境
LINUX_X86_64		Linux (x64)
MIPS_ABI		MIPS ABI
MVS_32	MVS	31 ビット SAS (z/OS)
MVS_64_BFP		64 ビット SAS (z/OS)
OS2		OS/2 (Intel)
RS_6000_AIX_32	RS_6000_AIX	AIX
RS_6000_AIX_64		AIX
SOLARIS_32	SOLARIS	Solaris (SPARC)
SOLARIS_64		Solaris (SPARC)
SOLARIS_X86_64		Solaris (x64)
VAX_VMS		OpenVMS VAX
VMS_IA64		OpenVMS (HP Integrity)
WINDOWS_32	windows	32 ビット SAS (Microsoft Windows)
WINDOWS_64		64 ビット SAS (Itanium ベースシステムと x64 両方の Microsoft Windows)

* 現在の値を使用することをお勧めします。エイリアスは互換性のためだけに提供されています。

詳細

注意:

エンコーディングに互換性がない場合、トランスコーディングで文字データが失われる可能性があります。エンコーディングおよびトランスコーディングの詳細については、*SAS 各国語サポート(NLS): リファレンスガイド*を参照してください。

関連項目:

- クロス環境データアクセス(CEDA)を使用したデータの処理
- “クロス環境データアクセス(CEDA)の定義”(SAS 言語リファレンス: 解説編 32 章)

POINTOBS=データセットオプション

SAS でオブザベーションへのランダムアクセスまたは順次アクセスが可能な圧縮データセットを作成するかどうかを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: オブザベーションコントロール

制限事項: POINTOBS=は、圧縮データセットを作成する場合にのみ適用されます。それ以外の場合は無視されます。

構文

POINTOBS=YES | NO

構文の説明

Yes

オブザベーション番号でランダムにアクセスできる圧縮データセットを SAS ソフトウェアで作成します。これがデフォルト設定です。

オブザベーション番号でデータに直接アクセスする例を次に示します。

- DATA ステップの MODIFY および SET ステートメントの POINT=オプション
- PROC FSEDIT を使用して特定のオブザベーション番号に直接アクセス

ヒント POINTOBS=YES と指定した場合、データセットからの情報の取得の効率性には影響しません。圧縮データセットの作成時およびその圧縮データセットの情報の更新または追加時に、CPU 使用率が約 10%増加します。

NO

オブザベーション番号で圧縮データセットのオブザベーションにランダムにアクセスできないようにします。

ヒント 圧縮データセット内の番号でオブザベーションを直接参照することが重要でないアプリケーションの場合、POINTOBS=NO を指定することをお勧めします。オブザベーション番号でデータにアクセスする必要がない場合、POINTOBS=NO と指定することで次の処理時に処理速度が約 10%向上します。

- 圧縮データセットの作成時
- 圧縮データセットのオブザベーションの更新または追加時

詳細

REUSE=YES は POINTOBS=YES よりも優先されます。次に、例を示します。

```
data test (compress=yes pointobs=yes reuse=yes);
```

この例では、データセットの設定は POINTOBS=NO になります。圧縮を使用する場合のデフォルトは POINTOBS=YES であるため、REUSE=YES によって POINTOBS=が NO に変更されます。

関連項目:

データセットオプション:

- [“COMPRESS=データセットオプション” \(13 ページ\)](#)
- [“REUSE=データセットオプション” \(53 ページ\)](#)

システムオプション:

- [“COMPRESS=システムオプション” \(SAS システムオプション: リファレンス\)](#)
- [“REUSE=システムオプション” \(SAS システムオプション: リファレンス\)](#)

PW=データセットオプション

SAS ファイルに READ、WRITE および ALTER パスワードを割り当て、パスワード保護された SAS ファイルへのアクセスを可能にします。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

構文

`PW=password`

構文の説明

password

パスワードを 8 文字に制限し、大文字と小文字を区別しない有効な SAS 名を指定する必要があります。“SAS 言語におけるワード” (SAS 言語リファレンス: 解説編 3 章) を参照してください。

詳細

PW=オプションは、カタログ以外のすべての種類の SAS ファイルに適用されます。このオプションは、パスワードを SAS ファイルに割り当てる場合や、パスワード保護された SAS ファイルにアクセスする場合に使用できます。

ALTER パスワードで保護された SAS データセットを置き換えると、新しいデータセットに ALTER パスワードが継承されます。新しいデータセットの ALTER パスワードを変更するには、DATASETS プロシジャで MODIFY ステートメントを使用します。

動作環境の情報

パスワードの使用の詳細については、動作環境に関する SAS のドキュメントの該当セクションを参照してください。

注: SAS パスワードでは、SAS システム以外での SAS ファイルへのアクセスはコントロールされません。SAS 以外で SAS ファイルへのアクセスをコントロールするには、オペレーティングシステムで提供されるユーティリティか、ファイルシステムのセキュリティコントロールを使用する必要があります。

関連項目:

データセットオプション:

- “ALTER=データセットオプション” (8 ページ)
- “ENCRYPT=データセットオプション” (17 ページ)
- “READ=データセットオプション” (48 ページ)
- “WRITE=データセットオプション” (69 ページ)

その他:

- “ファイルの保護” (SAS 言語リファレンス: 解説編 34 章)
- “パスワードの操作” (Base SAS プロシジャガイド 15 章)

PWREQ=データセットオプション

SAS データセットパスワードを入力するダイアログボックスを表示するかどうかを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

構文

PWREQ=YES | NO

構文の説明

Yes

ダイアログボックスを表示します。

NO

ダイアログボックスを表示しません。存在しないパスワードまたは無効なパスワードが入力された場合、データセットは開かずにエラーメッセージが SAS ログに書き込まれます。

詳細

対話型 SAS セッションでの PWREQ=オプションは、パスワード保護された SAS データセットに対してユーザーが無効なパスワードまたは存在しないパスワードを入力した後にダイアログボックスを表示するかどうかを制御します。PWREQ=は、READ、WRITE または ALTER パスワードのあるデータセットに適用されます。PWREQ=は SCL アプリケーションで最も役に立ちます。

関連項目:

データセットオプション:

- “ALTER=データセットオプション” (8 ページ)
- “ENCRYPT=データセットオプション” (17 ページ)
- “PW=データセットオプション” (47 ページ)
- “READ=データセットオプション” (48 ページ)
- “WRITE=データセットオプション” (69 ページ)

READ=データセットオプション

READ パスワードを SAS ファイルに割り当てます。これにより、ユーザーはパスワードを入力しない限りファイルを読み取れなくなります。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

構文

READ=*read-password*

構文の説明

read-password

有効な SAS 名にする必要があります。Rules for Words and Names in the SAS Language (SAS 言語リファレンス: 解説編) を参照してください。

詳細

READ=オプションは、カタログを以外のすべての種類の SAS ファイルに適用されません。このオプションは、パスワードを SAS ファイルに割り当てる場合や、読み取り保護された SAS ファイルにアクセスする場合に使用できます。

注: SAS パスワードでは、SAS システム以外での SAS ファイルへのアクセスはコントロールされません。SAS 以外で SAS ファイルへのアクセスをコントロールするには、オペレーティングシステムで提供されるユーティリティか、ファイルシステムのセキュリティコントロールを使用する必要があります。

関連項目:

データセットオプション:

- “ALTER=データセットオプション” (8 ページ)
- “ENCRYPT=データセットオプション” (17 ページ)
- “PW=データセットオプション” (47 ページ)
- “WRITE=データセットオプション” (69 ページ)

その他:

- “ファイルの保護” (SAS 言語リファレンス: 解説編 34 章)
- “パスワードの操作” (Base SAS プロシジャガイド 15 章)

RENAME=データセットオプション

変数の名前を変更します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: 変数制御

構文

RENAME=(*old-name-1=new-name-1 < ...old-name-n=new-name-n >*)

構文の説明

old-name

名前を変更する変数。

new-name

変数の新しい名前。有効な SAS 名を指定する必要があります。

詳細

データセットを作成するときに RENAME=データセットオプションを使用する場合は、出力データセットに新しい変数名を含める必要があります。入力データセットに対して RENAME=を使用すると、DATA ステップのプログラミングステートメントで新しい名前が使用されます。

SAS プロシジャで使用される入力データセットに対して RENAME=を使用すると、そのプロシジャ内の変数名が変更されます。WHERE ステートメントや WHERE=データセットオプションなどの WHERE 処理と一緒に RENAME=を使用すると、新しい名前はデータの処理前に適用されます。WHERE 式では新しい名前を使用する必要がありません。

同じ DATA ステップ内で DROP=または KEEP=データセットオプションのいずれかと RENAME=と一緒に使用すると、DROP=および KEEP=データセットオプションは RENAME=の前に適用されます。DROP=および KEEP=データセットオプションでは、古い名前を使用する必要があります。同じステートメント内で、同じ変数を削除したり名前変更したりすることはできません。

注: RENAME=データセットオプションは、出力モードで開くデータセットにのみ適用されます。

比較

- RENAME=データセットオプションと RENAME ステートメントには次の違いがあります。
 - RENAME=データセットオプションは PROC ステップ内で使用できますが、RENAME ステートメントは使用できません。
 - RENAME ステートメントは、すべての出力データセットに適用されます。異なるデータセットの異なる変数名を変更する場合は、RENAME=データセットオプションを使用する必要があります。
 - 処理が開始される前に変数名を変更するには、入力データセットに対して RENAME=データセットオプションを使用する必要があります。
- 同じ名前の変数が使用されている2つの入力データセットなど、プログラムロジックで変数名を変更する必要がある場合は、RENAME ステートメントまたは RENAME データセットオプションを使用します。ファイル管理タスクとして変数名を変更するには、DATASETS プロシジャを使用します。

例**例 1: 出力時の変数名の変更**

この例では、DATA ステートメントで RENAME=を使用し、出力データセットへの書き込み時に変数名が変更されることを示します。DATA ステップ処理時には、変数の元の名前 X が使用されます。

```
data one;
input x y z;
datalines;
24 595 439
243 343 034
;
```

例 2: 入力時の変数名の変更

この例では、SET ステートメントで変数 X を KEYS という変数名に変更します。この名前変更は DATA ステップ処理前に行われます。DATA ステップ処理の変数には、X ではなく、KEYS という名前が使用されます。

```
data three;
set one(rename=(x=keys));
z=keys+y;
run;
```

例 3: WHERE 処理で SAS プロシジャの変数名を変更する

この例では、PRINT プロシジャの変数 Score1 を Score2 という変数名に変更します。新しい名前はデータの処理前に適用されるため、WHERE ステートメントではこの新しい名前を指定する必要があります。

```
proc print data=test (rename=(score1=score2));
where score2 gt 75;
run;
```

関連項目:**データセットオプション:**

- [“DROP=データセットオプション” \(16 ページ\)](#)
- [“KEEP=データセットオプション” \(30 ページ\)](#)

ステートメント:

- [“RENAME ステートメント” \(SAS ステートメント: リファレンス\)](#)

プロシジャ:

- [“DATASETS プロシジャ” \(Base SAS プロシジャガイド\)](#)

REPEMPTY=データセットオプション

同じ名前の新しい空のデータセットで、既存の SAS データセットを上書きできるかどうかを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

制限事項: 出力データセットのみで使用。

構文

REPEMPTY=YES | NO

構文の説明**Yes**

同じ名前の新しい空のデータセットで、既存のデータセットを置き換えるように指定します。これがデフォルト設定です。

操作 REPEMPTY=YES および REPLACE=NO が設定されている場合、データセットは置き換えられません。

NO

同じ名前の新しい空のデータセットで、既存の SAS データセットを置き換えないように指定します。

ヒ 次の構文エラーで誤って作成された新しい空のデータセット B で既存のデータセット B が置き換えられないようにする場合、REPEMPTY=NO を使用します。

```
data mylib.a set b;
```

データを含む新しいデータセットで既存のデータセットを置き換え、誤って作成された新しい空のデータセットで既存のデータセットが上書きされないようにするには、REPLACE=YES および REPEMPTY=NO を設定します。

比較

- 個々のデータセットの場合、REPEMPTY=データセットオプションは LIBNAME ステートメントの REPEMPTY=オプションよりも優先されます。
- REPEMPTY=および REPLACE=データセットオプションは、永続的な SAS データセットと一時的な SAS データセットの両方に適用されます。ただし、REPLACE システムオプションは永続的な SAS データセットのみに適用されます。

関連項目:**データセットオプション:**

- [“REPLACE=データセットオプション” \(52 ページ\)](#)

ステートメントオプション:

- [“REPEMPTY=YES|NO” \(SAS ステートメント: リファレンス 2 章\)](#)

システムオプション:

- [“REPLACE システムオプション” \(SAS システムオプション: リファレンス\)](#)

REPLACE=データセットオプション

データが含まれている同じ名前の新しい SAS データセットで、既存のデータセットを上書きできるかどうかを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

制限事項: 出力データセットにのみ使用します。

このオプションは、SAS データセットの作成時にのみ有効です。

構文

REPLACE=NO | YES

構文の説明

NO

同じ名前の新しいデータセットで、既存のデータセットを置き換えないように指定します。

Yes

同じ名前の新しいデータセットで、既存のデータセットを置き換えるように指定します。

詳細

- REPLACE=データセットオプションは、個々のデータセットの REPLACE システムオプションよりも優先されます。
- REPLACE システムオプションは、永続的な SAS データセットにのみ適用されます。

例

この DATA ステートメントで REPLACE=データセットオプションを使用すると、MYLIB によって参照されるライブラリ内の ONE という名前の永続的な SAS データセットは置き換えられなくなります。

```
data mylib.one(replace=no);
```

ファイルが置き換えられていないことを示すメッセージがログに書き込まれます。

関連項目:

システムオプション:

- “REPLACE システムオプション” (SAS システムオプション: リファレンス)

REUSE=データセットオプション

圧縮 SAS データセットの空き領域に新しいオブザベーションを書き込みできるかどうかを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

制限事項: 出力データセットにのみ使用します。

構文

REUSE=NO | YES

構文の説明

NO

圧縮データセット内の領域を追跡および再利用しません。新しいオブザベーションは、既存のデータセットの最後に追加されます。SAS データセット内の多数のオブザベーションを削除したり更新したりする場合には、NO 引数を指定するとデータの格納効率が下がります。

Yes

圧縮 SAS データセット内の領域を追跡して再利用します。新しいオブザベーションは、他のオブザベーションの更新または削除によって空いた領域に挿入されます。

SAS データセットの末尾にオブザベーションを追加するプロシジャ (APPEND プロシジャや FSEDIT プロシジャなど) を圧縮データセットで使用する場合は、REUSE=NO 引数を使用します。REUSE=YES を指定すると、新しいオブザベーションは必ずしも末尾ではない、ファイル内の領域に追加されます。

詳細

デフォルトでは、新しいオブザベーションは既存の圧縮データセットの最後に追加されます。他のオブザベーションの削除または更新によって空いた領域を追跡し、再利用する場合は、圧縮 SAS データセットを作成するときに REUSE=データセットオプションを使用します。

REUSE=の使用は、COMPRESS=YES データセットオプションまたはシステムオプションを使用してデータセットを新規作成するときのみ意味があります。既存の SAS データセットにアクセスするときに REUSE=データセットオプションを使用しても効果はありません。

比較

REUSE=データセットオプションは、REUSE=システムオプションよりも優先されます。

REUSE=YES は、POINTOBS=YES よりも優先されます。たとえば、次のステートメントを使用すると、データセットの設定は POINTOBS=NO になります。

```
data test (compress=yes pointobs=yes reuse=yes);
```

圧縮を使用する場合のデフォルトは POINTOBS=YES であるため、REUSE=YES によって POINTOBS=が NO に変更されます。

関連項目:**データセットオプション:**

- [“COMPRESS=データセットオプション” \(13 ページ\)](#)

システムオプション:

- [“REUSE=システムオプション” \(SAS システムオプション: リファレンス\)](#)

ROLE=データセットオプション

スタースキーマ結合のファクトテーブルを指定します。

該当要素: PROC SQL

カテゴリ: データセットコントロール

制限事項: 入力データセットにのみ使用します。

構文

ROLE= FACT | DIMENSION | DIM

構文の説明

FACT

SAS データセットをスタースキーマのファクトテーブルとして指定します。

DIMENSION | DIM

SAS データセットをスタースキーマのディメンションテーブルとして指定します。

詳細

スタースキーマは複数のテーブルの配列で、大きなファクトテーブルが複数のディメンションテーブルに結合されます。たとえば、スタースキーマを作成するために、SQL プロシジャの構文を使用して SAS データセットを結合できます。

結合テーブルを処理するアプリケーションの処理速度を向上するには、ROLE=データセットオプションを指定します。たとえば、ROLE=FACT を指定して特定のファクトテーブルを指定できます。または、ROLE=DIMENSION を指定して各ディメンションテーブルを指定できます。この場合は未指定のテーブルがファクトテーブルということになります。

テーブルの役割はクエリごとに変わる可能性があるため、ROLE=の指定は現在のステップにのみ有効で、データセットには保存されません。

例: ファクトテーブルの指定

次の例では、ROLE=データセットオプションの使用により、PROC SQL の処理速度が向上します。ORDERS がファクトテーブルで、PRODUCT、PERIOD、CUSTOMER がディメンションテーブルです。

```
proc sql;
select orders.Order_Total
from orders (role=fact), product, period, customer
where orders.Product_ID = product.Product_ID
and orders.Period_ID = period.Period_ID
and orders.Customer_ID = customer.Customer_ID
and product.Product_Name = "camera"
and period.Period_Name = "1997"
and customer.Customer_Name = "Walmart";
quit;
```

関連項目:

“SQL プロシジャ” (SAS SQL プロシジャユーザーガイド)

SORTEDBY=データセットオプション

データセットの現在の並べ替え方法を指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

構文

SORTEDBY=*by-clause* </ *collate-name*> | **_NULL_**

構文の説明

by-clause < / *collate-name* >

データの現在の並べ替え方法を示します。

by-clause

PROC SORT ステップの BY ステートメントで使用する変数とオプションを指定します。

collate-name

並べ替えに使用する照合順序の名前を指定します。デフォルトでは、動作環境の照合順序になります。照合順序の前にスラッシュ(/)を付ける必要があります。

動作環境の情報

照合順序の詳細については、動作環境に関する SAS のドキュメントを参照してください。

NULL

既存の並べ替えインジケータを削除します。

詳細

並べ替えインジケータを確認することにより、データセットがすでに key 変数に基づき昇順に並べ替えられているかどうか判断されます。並べ替えインジケータは、データセットのディスクリプタ情報に保存され、以前の並べ替えから設定されます。並べ替えインジケータの使用法と処理速度向上の詳細については、“ソートインジケータ”(SAS 言語リファレンス: 解説編 25 章) および“SORTVALIDATE システムオプション”(SAS システムオプション: リファレンス) を参照してください。

次の例は、Validated 属性が NO に設定された、CONTENTS プロシジャの Sort Information セクションで、SORTEDBY=データセットオプションを使用してデータセットが並べ替えられたことを示しています。

```
Sort Information
Sortedby var1
Validated NO
Character Set ANSI
```

比較

- データセットの並べ替え方法を表示するには、DATASETS プロシジャで CONTENTS ステートメントを使用します。
- SORTEDBY=オプションは、データの並べ替え方法を示しますが、データセットの並べ替えは実行しません。

例

この例では、SORTEDBY=データセットオプションを使用して、データの現在の並べ替え方法を指定します。データセット ORDERS は、PRIORITY に基づき、INDATE 値の降順に並べ替えられます。データセットが作成されると、並べ替えインジケータと一緒に保存されます。これらのステートメントにより、データセット ORDERS が作成され、並べ替えインジケータが記録されます。

```
libname mylib 'SAS-library';
options yearcutoff=1920;
data mylib.orders(sortedby=priority
descending indate);
input priority 1. +1 indate date7.
```

```

+1 office $ code $;
format indate date7.;
datalines;
1 03may01 CH J8U
1 21mar01 LA M91
1 01dec00 FW L6R
1 27feb99 FW Q2A
2 15jan00 FW I9U
2 09jul99 CH P3Q
3 08apr99 CH H5T
3 31jan99 FW D2W
;

```

関連項目:

- “DATASETS プロシジャ” (Base SAS プロシジャガイド)
- “SORT プロシジャ” (Base SAS プロシジャガイド)
- “SQL プロシジャ” (SAS SQL プロシジャユーザーガイド)

SPILL=データセットオプション

DATA ステップビューの非順次処理の予備ファイルを作成するかどうかを指定します。

- 該当要素:** DATA ステップおよび PROC ステップ
- カテゴリ:** データセットコントロール
- 制限事項:** DATA ステップビューでのみ有効です。
-

構文

SPILL=YES | NO

構文の説明

Yes

DATA ステップビューの非順次処理の予備ファイルを作成します。これがデフォルト設定です。

操作 DATA ステップビューの順次処理に予備ファイルが作成されることはありません。

NO

予備ファイルを作成しないか、予備ファイルのサイズを小さくします。

操作 直接(ランダム)アクセスの場合は、SPILL=NO を指定しても必ず予備ファイルが作成されます。

注 大量のデータを生成する DATA ステップビューの予備ファイルに対応する十分なディスク容量がない場合は、SPILL=NO を指定してください。

ヒント BY グループデータを処理する SAS プロシジャの場合は、現在の BY グループのみが予備ファイルに書き込まれるように SPILL=NO を指定することを検討してください。

詳細

DATA ステップビューが非順次処理で開かれるときに、デフォルトで予備ファイルが作成されます。*spill file* には、DATA ステップビューによって生成されるオブザベーションが含まれます。以降のデータ要求では、DATA ステップビューを再び実行するか代わりに、予備ファイルからオブザベーションが読み込まれます。予備ファイルは WORK ライブラリにある一時ファイルです。

非順次処理には、一部の SAS ステートメントおよびプロシジャでサポートされる、次のアクセスメソッドが含まれます。SPILL=データセットオプションが各アクセスメソッドでどのように動作するかを次に説明します。

ランダムアクセス

すべてのオブザベーションを順次に読み込まずにインデックスを使用して、オブザベーション番号か変数の値のいずれかで、オブザベーションを直接取得します。オブザベーションごとに DATA ステップビューを再起動すると処理時間が非常に長くなるため、SPILL=YES と SPILL=NO のどちらであっても、予備ファイルは必ず作成されます。

BY-グループアクセス

BY ステートメントを使用して、変数の値に基づいて順序付け、グループ化またはインデックス付けされるオブザベーションを処理します。SPILL=YES の場合、DATA ステップビューから要求されるすべてのデータと同じサイズの予備ファイルが作成されます。SPILL=NO の場合、現在の BY グループのみを予備ファイルに書き込みます。予備ファイルの最大サイズは、最大の BY グループを格納できるサイズです。

2 パスアクセス

データのパススルーを順次に複数回実行します。SPILL=NO の場合、予備ファイルは作成されません。かわりに、データの 1 回目のパススルーの後、後続のパススルーごとに DATA ステップビューが再起動されます。再起動ごとに DATA ステップビューによって少量のデータが返されると、ビューの再起動処理時間による影響が著しくなる可能性があります。

注: SPILL=NO では、データの後続のパススルーで、異なるデータが生成される場合があります。予備ファイルの使用が必要になる処理もあります。たとえば、ランダム関数を使用した現在時刻に基づく値の計算結果がデータに影響する可能性があります。

例

例 1: 少数の大きな BY グループに予備ファイルを使用する

この例では、大量のランダムデータを生成し、BY ステートメントで UNIVARIATE プロシジャを使用する、DATA ステップビューを作成します。この例は、少数の大きな BY グループで SPILL=を使用した場合の影響を示します。

SPILL=YES の場合、DATA ステップビューから要求されるすべてのオブザベーションが予備ファイルに書き込まれます。SPILL=NO の場合、現在の BY グループにあるオブザベーションのみが予備ファイルに書き込まれます。この例によって生成される通知メッセージは、SPILL=NO で縮小される予備ファイルのサイズを示します。ただし、BY グループごとに予備ファイルを切り捨てる時間により、DATA ステップビューの全体的な処理時間が増えることがあります。

```
options msglevel=i;
data vw_few_large / view=vw_few_large;
drop i;
do byval = 'Group A', 'Group B', 'Group C';
do i = 1 to 500000;
```

```
r = ranuni(4);  
output;  
end;  
end;  
run;  
proc univariate data=vw_few_large (spill=yes) noprint;  
var r;  
by byval;  
run;  
proc univariate data=vw_few_large (spill=no) noprint;  
var r;  
by byval;  
run;
```

ログ2.1 SAS ログ出力

```

1 options msglevel=i;
2 data vw_few_large / view=vw_few_large;
3 drop i;
4
5 do byval = 'Group A', 'Group B', 'Group C';
6 do i = 1 to 500000;
7 r = ranuni(4);
8 output;
9 end;
10 end;
11 run;
NOTE: DATA STEP view saved on file WORK.VW_FEW_LARGE.
NOTE: A stored DATA STEP view cannot run under a different operating system.
NOTE: DATA statement used (Total process time):
real time 21.57 seconds
cpu time 1.31 seconds
12 proc univariate data=vw_few_large (spill=yes) noprint;
INFO: View WORK.VW_FEW_LARGE open mode: BY-group rewind.
13 var r;
14 by byval;
15 run;
INFO: View WORK.VW_FEW_LARGE opening spill file for output observations.
INFO: View WORK.VW_FEW_LARGE deleting spill file. File size was 22506120 bytes.
NOTE: View WORK.VW_FEW_LARGE.VIEW used (Total process time):
real time 40.68 seconds
cpu time 12.71 seconds
NOTE: PROCEDURE UNIVARIATE used (Total process time):
real time 57.63 seconds
cpu time 13.12 seconds
16
17 proc univariate data=vw_few_large (spill=no) noprint;
INFO: View WORK.VW_FEW_LARGE open mode: BY-group rewind.
18 var r;
19 by byval;
20 run;
INFO: View WORK.VW_FEW_LARGE opening spill file for output observations.
INFO: View WORK.VW_FEW_LARGE truncating spill file. File size was 7502040 bytes.
NOTE: The above message was for the following by-group:
byval=Group A
INFO: View WORK.VW_FEW_LARGE truncating spill file. File size was 7534800 bytes.
NOTE: The above message was for the following by-group:
byval=Group B
INFO: View WORK.VW_FEW_LARGE truncating spill file. File size was 7534800 bytes.
NOTE: The above message was for the following by-group:
byval=Group C
INFO: View WORK.VW_FEW_LARGE deleting spill file. File size was 32760 bytes.
NOTE: View WORK.VW_FEW_LARGE.VIEW used (Total process time):
real time 11.03 seconds
cpu time 10.95 seconds
NOTE: PROCEDURE UNIVARIATE used (Total process time):
real time 11.04 seconds
cpu time 10.96 seconds

```

例 2: 多数の小さな BY グループに予備ファイルを使用する

この例では、大量のランダムデータを生成し、BY ステートメントで UNIVARIATE プロシジャを使用する、DATA ステップビューを作成します。この例は、多数の小さな BY グループで SPILL=を使用した場合の影響を示します。

SPILL=YES の場合、DATA ステップビューから要求されるすべてのオブザベーションが予備ファイルに書き込まれます。SPILL=NO の場合、現在の BY グループにあるオブザベーションのみが予備ファイルに書き込まれます。この例によって生成される通知

メッセージは、SPILL=NO で縮小される予備ファイルのサイズを示します。小さな BY グループの場合は、大量のディスク容量を節約できます。

```
options msglevel=i;
data vw_many_small / view=vw_many_small;
drop i;
do byval = 1 to 100000;
do i = 1 to 5;
r = ranuni(4);
output;
end;
end;
run;
proc univariate data=vw_many_small (spill=yes) noprint;
var r;
by byval;
run;
proc univariate data=vw_many_small (spill=no) noprint;
var r;
by byval;
run;
```

ログ2.2 SAS ログ出力

```
1 options msglevel=i;
2 data vw_many_small / view=vw_many_small;
3 drop i;
4
5 do byval = 1 to 100000;
6 do i = 1 to 5;
7 r = ranuni(4);
8 output;
9 end;
10 end;
11 run;
NOTE: DATA STEP view saved on file WORK.VW_MANY_SMALL.
NOTE: A stored DATA STEP view cannot run under a different operating system.
NOTE: DATA statement used (Total process time):
real time 0.56 seconds
cpu time 0.03 seconds
12 proc univariate data=vw_many_small (spill=yes) noprint;
INFO: View WORK.VW_MANY_SMALL open mode: BY-group rewind.
13 var r;
14 by byval;
15 run;
INFO: View WORK.VW_MANY_SMALL opening spill file for output observations.
INFO: View WORK.VW_MANY_SMALL deleting spill file. File size was 8024240 bytes.
NOTE: View WORK.VW_MANY_SMALL.VIEW used (Total process time):
real time 30.73 seconds
cpu time 29.59 seconds
NOTE: PROCEDURE UNIVARIATE used (Total process time):
real time 30.96 seconds
cpu time 29.68 seconds
16
17 proc univariate data=vw_many_small (spill=no) noprint;
INFO: View WORK.VW_MANY_SMALL open mode: BY-group rewind.
18 var r;
19 by byval;
20 run;
INFO: View WORK.VW_MANY_SMALL opening spill file for output observations.
INFO: View WORK.VW_MANY_SMALL truncating spill file. File size was 65504 bytes.
NOTE: The above message was for the following by-group:
byval=410
INFO: View WORK.VW_MANY_SMALL truncating spill file. File size was 65504 bytes.
NOTE: The above message was for the following by-group:
byval=819
INFO: View WORK.VW_MANY_SMALL truncating spill file. File size was 65504 bytes.
NOTE: The above message was for the following by-group:
byval=1229

.
. Deleted many INFO and NOTE messages for BY groups
.
INFO: View WORK.VW_MANY_SMALL truncating spill file. File size was 65504 bytes.
NOTE: The above message was for the following by-group:
byval=99894
INFO: View WORK.VW_MANY_SMALL deleting spill file. File size was 32752 bytes.
NOTE: View WORK.VW_MANY_SMALL.VIEW used (Total process time):
real time 29.43 seconds
cpu time 28.81 seconds
NOTE: PROCEDURE UNIVARIATE used (Total process time):
real time 29.43 seconds
cpu time 28.81 seconds
```

例 3: 2 パスアクセスで予備ファイルを使用する

この例では、大量のランダムデータを生成し、TRANSPOSE プロシジャを使用する、DATA ステップビューを作成します。この例は、2 パスアクセス処理が必要なプロシジャで SPILL=を使用した場合の影響を示します。

PROC TRANSPOSE が DATA ステップビューを処理するとき、プロシジャはビューが生成するオブザベーションを 2 回パススルーする必要があります。1 回目のパスでオブザベーション数を数え、2 回目のパスで転置が実行されます。SPILL=YES の場合、1 回目のパス時に予備ファイルが作成され、2 回目のパスで予備ファイルからオブザベーションが読み込まれます。SPILL=NO の場合、予備ファイルは作成されず、1 回目のパスの後に DATA ステップビューが再起動されます。

SPILL=データセットオプションを含まない、最初の TRANSPOSE プロシジャでは、デフォルトで予備ファイルが使用される場合でも、open モードに関する通知メッセージは表示されません。このアクションは、SPILL=データセットオプションを使用していないユーザーに対する SAS ログのメッセージ量を減らすために行われます。

```
options msglevel=i;
data vw_transpose/view=vw_transpose;
drop i j;
array x[10000];
do i = 1 to 10;
do j = 1 to dim(x);
x[j] = ranuni(4);
end;
output;
end;
run;
proc transpose data=vw_transpose out=transposed;
run;
proc transpose data=vw_transpose (spill=yes) out=transposed;
run;
proc transpose data=vw_transpose (spill=no) out=transposed;
run;
```

ログ2.3 SAS ログ出力

```
1 options msglevel=i;
2 data vw_transpose/view=vw_transpose;
3 drop i j;
4 array x[10000];
5 do i = 1 to 10;
6 do j = 1 to dim(x);
7 x[j] = ranuni(4);
8 end;
9 output;
10 end;
11 run;
NOTE: DATA STEP view saved on file WORK.VW_TRANSPOSE.
NOTE: A stored DATA STEP view cannot run under a different operating system.
NOTE: DATA statement used (Total process time):
real time 0.68 seconds
cpu time 0.18 seconds
12 proc transpose data=vw_transpose out=transposed;
13 run;
INFO: View WORK.VW_TRANSPOSE opening spill file for output observations.
INFO: View WORK.VW_TRANSPOSE deleting spill file. File size was 880000 bytes.
NOTE: View WORK.VW_TRANSPOSE.VIEW used (Total process time):
real time 2.37 seconds
cpu time 1.17 seconds
NOTE: There were 10 observations read from the data set WORK.VW_TRANSPOSE.
NOTE: The data set WORK.TRANSPOSED has 10000 observations and 11 variables.
NOTE: PROCEDURE TRANSPOSE used (Total process time):
real time 4.17 seconds
cpu time 1.51 seconds
14 proc transpose data=vw_transpose (spill=yes) out=transposed;
INFO: View WORK.VW_TRANSPOSE open mode: sequential.
15 run;
INFO: View WORK.VW_TRANSPOSE reopen mode: two-pass.
INFO: View WORK.VW_TRANSPOSE opening spill file for output observations.
INFO: View WORK.VW_TRANSPOSE deleting spill file. File size was 880000 bytes.
NOTE: View WORK.VW_TRANSPOSE.VIEW used (Total process time):
real time 0.95 seconds
cpu time 0.92 seconds
NOTE: There were 10 observations read from the data set WORK.VW_TRANSPOSE.
NOTE: The data set WORK.TRANSPOSED has 10000 observations and 11 variables.
NOTE: PROCEDURE TRANSPOSE used (Total process time):
real time 1.01 seconds
cpu time 0.98 seconds
16 proc transpose data=vw_transpose (spill=no) out=transposed;
INFO: View WORK.VW_TRANSPOSE open mode: sequential.
17 run;
INFO: View WORK.VW_TRANSPOSE reopen mode: two-pass.
INFO: View WORK.VW_TRANSPOSE restarting for another pass through the data.
NOTE: View WORK.VW_TRANSPOSE.VIEW used (Total process time):
real time 1.34 seconds
cpu time 1.32 seconds
NOTE: The View WORK.VW_TRANSPOSE was restarted 1 times. The following view
statistics
only apply to the last view restart.
NOTE: There were 10 observations read from the data set WORK.VW_TRANSPOSE.
NOTE: The data set WORK.TRANSPOSED has 10000 observations and 11 variables.
NOTE: PROCEDURE TRANSPOSE used (Total process time):
real time 1.42 seconds
cpu time 1.40 seconds
```

関連項目:

データセットオプション:

- “OBSBUF=データセットオプション” (33 ページ)

TOBSNO=データセットオプション

クライアント/サーバー間の転送で送信するオブザベーション数を指定します。

- 該当要素:** DATA ステップおよび PROC ステップ
- カテゴリ:** データセットコントロール
- 制限事項:** TOBSNO=オプションは、REMOTE エンジンを経由して SAS サーバーからアクセスするデータセットでのみ有効です。

構文

TOBSNO=*n*

構文の説明

n
送信するオブザベーション数を指定します。

詳細

TOBSNO=オプションが指定されていない場合、オブザベーションの長さ、PROC SERVER ステートメントの TBUFSIZE=オプションで指定されたサーバーの送信バッファサイズに基づいて値が計算されます。

TOBSNO=オプションは、REMOTE エンジンを経由して SAS サーバーからアクセスするデータセットでのみ有効です。更新するために開かれるデータセット、または別のエンジン経由でアクセスしたデータセットにこのオプションが指定されている場合は無視されます。

関連項目:

“FOPEN 関数” (SAS 関数と CALL ルーチン: リファレンス)

TYPE=データセットオプション

特別構造の SAS データセットにデータセットの種類を指定します。

- 該当要素:** DATA ステップおよび PROC ステップ
- カテゴリ:** データセットコントロール

構文

TYPE=*data-set-type*

構文の説明

data-set-type
特殊なデータセットの種類を指定します。

詳細

適切な出力形式で特殊な SAS データセットを作成する、またはプロシジャステートメント内の特殊な SAS データセットの種類を指定するには、DATA ステップで TYPE=データセットオプションを使用します。

データセットの種類を判断するには、CONTENTS プロシジャを使用できます。

ほとんどの SAS データセットには、指定種類がありません。ただし、一部の SAS/STAT プロシジャで使用される、特別構造の SAS データセットがいくつかあります。これらの SAS データセットには、特殊な変数とオブザベーションが含まれており、これらは通常 SAS 統計プロシジャで作成されます。特殊な SAS データセットのほとんどは SAS/STAT ソフトウェアで使用されるため、これらの詳細については *SAS/STAT User's Guide* を参照してください。特殊データセットには、CORR、COV、SSPC、EST、FACTOR などがあります。

その他に、他の SAS ソフトウェア製品で使用できる値があります。詳細については、適切な製品のドキュメントを参照してください。

注: DATA ステップと SET ステートメントを一緒に使用して特殊な SAS データセットを変更する場合は、DATA ステートメントで TYPE=オプションを指定する必要があります。data-set-type は作成するデータセットには自動的にコピーされません。

関連項目:

- “CONTENTS プロシジャ” (Base SAS プロシジャガイド)

WHERE=データセットオプション

SAS データセットからオブザベーションを選択するために使用する特定の条件を指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: オブザベーションコントロール

制限事項: SET および MODIFY ステートメント内で POINT=オプションと一緒に使用できません。

構文

WHERE=(*where-expression-1*<*logical-operator* *where-expression-n*>)

構文の説明

where-expression

一連の演算子、オペランドおよび SAS 関数で構成される演算式または論理式です。オペランドは変数、SAS 関数または定数です。演算子は、比較、論理演算または算術計算を要求する記号です。式はかっこで囲む必要があります。

logical-operator

AND、AND NOT、OR、OR NOT のいずれかを指定できます。

詳細

- 処理のために DATA ステップまたは PROC ステップに渡される前に、WHERE 式で指定した条件を満たすオブザベーションを選択するには、入力データセットで WHERE=データオプションを使用します。WHERE 式の条件を満たすオブザベーションの選択は、DATA ステップの各反復で最初に行われる操作です。

また、出力データセットに書き込まれるオブザベーションを選択することもできます。一般的には、入力時点でオブザベーションを選択するほうが、出力時点で選択するより効率的です。ただし、入力時点でのオブザベーションの選択が実用的でない場合や不可能な場合があります。

- WHERE 処理には、OBS=処理および FIRSTOBS=処理を適用できます。詳細については、“条件選択によるサブセットデータの処理” (SAS 言語リファレンス: 解説編 11 章) を参照してください。
- SET および MODIFY ステートメント内で、WHERE=データセットオプションと POINT=オプションを一緒に使用することはできません。
- 同じ DATA ステップで WHERE=データセットオプションと WHERE ステートメントの両方を使用すると、WHERE データセットオプションを使用するデータセットの WHERE ステートメントは無視されます。ただし、SAS/FSP ソフトウェアでは、WHERE=データセットオプションと WHERE コマンドを一緒に使用できます。

注: インデックス付き SAS データセットを使用すると、WHERE 式を使用して SAS データセット内のオブザベーションのサブセットにアクセスするときに、処理速度を大幅に向上できます。インデックス付きデータセットでの WHERE 式の処理の詳細、および SAS データセットにインデックスを付ける前に検討するガイドラインの一覧については、“SAS インデックスについて” (SAS 言語リファレンス: 解説編 26 章) を参照してください。

比較

- WHERE ステートメントはすべての入力データセットに適用されますが、WHERE=データセットオプションは指定されたデータセットのみからオブザベーションを選択します。
- WHERE=データセットオプションの目的を混同しないようにしてください。DROP=および KEEP=データセットオプションでは処理対象の変数が選択されますが、WHERE=データセットオプションではオブザベーションが選択されます。

例

例 1: 入力データセットからオブザベーションを選択する

この例では、WHERE=データセットオプションを使用して、別のデータセットに読み込まれるときに SALES データセットをサブセット化します。

```
data whizmo; set sales(where=(product='whizmo')); run;
```

例 2: 出力データセットからオブザベーションを選択する

この例では、WHERE=データセットオプションを使用して、SALES 出力データセットをサブセット化します。

```
data whizmo(where=(product='whizmo')); set sales; run;
```

関連項目:

- “WHERE ステートメント” (SAS ステートメント: リファレンス)
- WHERE 式の処理

WHEREUP=データセットオプション

新しいオブザベーションと変更されたオブザベーションを WHERE 式に対して評価するかどうかを指定します。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: オブザベーションコントロール

構文

WHEREUP=NO | YES

構文の説明

NO

追加されたオブザベーションと変更されたオブザベーションは WHERE 式に対して評価しません。

Yes

追加されたオブザベーションと変更されたオブザベーションを WHERE 式に対して評価します。

詳細

追加されたオブザベーションまたは変更されたオブザベーションを指定した WHERE 式と照合する場合、WHEREUP=YES を指定します。

例

例 1: WHERE 式に一致しない更新の受け入れ

この例では、変更されたオブザベーションが WHERE 式に一致しない場合でもオブザベーションの更新および追加を WHEREUP=で許可する方法を示します。

```
data a;
x=1;
output;
x=2;
output;
run;
data a;
modify a(where=(x=1) whereup=no);
x=3;
replace; /* Update does not match WHERE expression */
output; /* Add does not match WHERE expression */
run;
```

この例では、SAS によってオブザベーションが更新され、新しいオブザベーションがデータセットに追加されます。

例 2: WHERE 式に一致しない更新の拒否

この例では、更新および追加が WHERE 式に一致しない場合、WHEREUP=はオブザベーションの更新または追加を許可しません。

```
data a;
```



```

x=1;
output;
x=2;
output;
run;
data a;
modify a(where=(x=1) whereup=yes);
x=3;
replace; /* Update does not match WHERE expression */
output; /* Add does not match WHERE expression */
run;

```

この例では、SAS によってオブザベーションは更新されず、新しいオブザベーションはデータセットに追加されません。

関連項目:

データセットオプション:

- “WHERE=データセットオプション” (66 ページ)

WRITE=データセットオプション

WRITE パスワードを SAS ファイルに割り当てます。これにより、ユーザーはパスワードを入力しない限りファイルを書き込めなくなります。

該当要素: DATA ステップおよび PROC ステップ

カテゴリ: データセットコントロール

構文

WRITE=*write-password*

構文の説明

write-password

有効な SAS 名にする必要があります。

参照項目 Rules for Words and Names in the SAS Language (*SAS 言語リファレンス: 解説編*)

詳細

WRITE=オプションは、カタログ以外のすべての種類の SAS ファイルに適用されます。このオプションは、パスワードを SAS ファイルに割り当てる場合や、書き込み保護された SAS ファイルにアクセスする場合に使用できます。

注: SAS パスワードでは、SAS システム以外での SAS ファイルへのアクセスはコントロールされません。SAS 以外で SAS ファイルへのアクセスをコントロールするには、オペレーティングシステムで提供されるユーティリティか、ファイルシステムのセキュリティコントロールを使用する必要があります。

関連項目:

データセットオプション:

- “ALTER=データセットオプション” (8 ページ)
- “ENCRYPT=データセットオプション” (17 ページ)
- “PW=データセットオプション” (47 ページ)
- “READ=データセットオプション” (48 ページ)

その他:

- “ファイルの保護” (SAS 言語リファレンス: 解説編 34 章)
- “パスワードの操作” (Base SAS プロシジャガイド 15 章)

キーワード

2

2 パスアクセス
予備ファイル 58

A

ALTER=データセットオプション 8
ALTER パスワード
SAS ファイルへの割り当て 47

B

BUFNO=データセットオプション 9
BUFSIZE=データセットオプション 11
BY-グループアクセス
予備ファイル 58

C

CNTLLEV=データセットオプション 12
COMPRESS=データセットオプション 13

D

DATA ステップビュー
非順次処理の予備ファイル 57
表示バッファサイズ 33
DLDMGACTION=データセットオプション 15
DROP=データセットオプション 16

E

ENCRYPT=データセットオプション 17
EXTENDOBSCOUNTER=データセット
オプション 19

F

FILECLOSE=データセットオプション 20
FIRSTOBS=データセットオプション 20

G

GENMAX=データセットオプション 22
GENNUM=データセットオプション 23

I

IDXNAME=データセットオプション 25
IDXWHERE=データセットオプション 26
IN=データセットオプション 29
INDEX=データセットオプション 28

K

KEEP=データセットオプション 30

L

LABEL=データセットオプション 31

O

OBS=データセットオプション
WHERE 処理 37
オブザベーションが削除されたデータ
セット 40
比較 36
OBS=データセットオプション
例 35
OBSBUF=データセットオプション 33
OUTREP=データセットオプション 44

P

POINTOBS=データセットオプション 45
PW=データセットオプション 47
PWREQ=データセットオプション 48

R

RDC (Ross Data Compression) 13
READ=データセットオプション 48

READ パスワード

SAS ファイルへの割り当て 47, 48

RENAME=データセットオプション 49

REPEMPTY=データセットオプション 51

REPLACE=データセットオプション 52

REUSE=データセットオプション 53

RLE (Run Length Encoding) 13

ROLE=データセットオプション 54

Ross Data Compression (RDC) 13

Run Length Encoding (RLE) 13

S

SAS ファイル

ALTER パスワード 8

書き込みを禁止する 69

パスワードの割り当て 47

読み込みを禁止する 48

SORTEDBY=データセットオプション 55

SPILL=データセットオプション 57

T

TOBSNO=データセットオプション 65

TYPE=データセットオプション 65

W

WHERE=データセットオプション 66

WHEREUP=データセットオプション 68

WHERE 式

インデックス検索 26

インデックスの無効化 26

更新されたデータセットの評価 68

順次検索 26

条件に一致するインデックスを指定する 25

WHERE の処理

OBS=データセットオプション 37

プロシジャの変数名を変更する 51

WRITE=データセットオプション 69

WRITE パスワード

SAS ファイルへの割り当て 47, 69

あ

アクセスメソッド

非順次処理 58

圧縮データセット

空き領域の再利用 53

暗号化

出力データセット 17

インデックス

インデックス検索の指定 26

候補インデックスの指定 25

出力データセットの定義 28

損傷したデータセット 15

無効化 26

永久バッファページ

サイズ 11

オブザベーション

1つのデータセットで最初に処理するオブザベーション 20

空き領域への書き込み 53

カウントの拡張 19

クライアント/サーバー間の転送 65

更新されたデータセットと WHERE 式 68

出力データセットの圧縮 13

処理の終点 35

選択条件 66

データセットの関与 29

オブザベーションカウント

拡張 19

か

空のデータセット 51

共用アクセスレベル

データセット 12

クライアント/サーバー間の転送

送信するオブザベーションの数 65

結合

スタースキーマ結合のファクトテーブル 54

さ

システムオプション

データセットの相互作用 3

出力データセット

暗号化 17

インデックスの定義 28

永久バッファのページサイズ 11

オブザベーションカウントの拡張 19

オブザベーションの圧縮 13

オブザベーションの選択 67

書き込みから変数を除外する 16

書き込む変数の指定 30

データセットオプション 2

データ表現 44

スタースキーマ結合

ファクトテーブル 54

世代

数の変更 22

データセットの指定 23

データセットの要求 22

バージョンの最大数 22

ソート

データセットの並べ替え情報 55

損傷したデータセット 15

た

- ダイアログボックス
 - データセットパスワードの入力 48
- データセット
 - 1つのデータセットで最初に処理するオブザベーション 20
 - 暗号化 17
 - 上書き 51, 52
 - オブザベーションカウントの拡張 19
 - オブザベーションの選択 66
 - オブザベーションの選択条件 66
 - 空にする 51
 - 共用アクセスレベル 12
 - 現在のオブザベーションへの関与 29
 - 更新, WHERE 式の評価 68
 - 最後に処理するオブザベーション 35
 - 修理 15
 - 終了時のテープボリューム位置 20
 - 処理に割り当てられるバッファ 9
 - 世代 22
 - 世代, 指定 23
 - ソート 55
 - 損傷 15
 - 置換 52
 - 同名 51, 52
 - 特別構造 65
 - パスワード入力ダイアログボックス 48
 - 変数の削除 16
 - 変数の保持 30
 - ラベル 31
 - 取り除き 13
- データセットオプション 1
 - 構文 2
 - システムオプションの相互作用 3
 - 出力データセット 2
 - 入力データセット 2
 - 例 2
- データセットタイプ
 - 特別構造のデータセット 65
- データセットの圧縮 13
 - ランダムアクセスと順次アクセス 45
- データセットの修復 15
- データ表現
 - 出力データセット 44
- テープボリューム
 - データセット終了時の位置 20

な

- 並べ替え情報
 - データセット 55

入力データセット

- オブザベーションの選択 67
- 処理から変数を除外する 16
- 処理する変数の指定 30
- データセットオプション 2

は

- パスワード
 - ALTER パスワード 8
 - READ パスワード 48
 - SAS ファイルへの割り当て 47
 - WRITE パスワード 69
 - 入力ダイアログボックス 48
- パスワード保護されたファイル
 - アクセスの有効化 47
- バッファ
 - 永久バッファのページサイズ 11
 - データセット処理での割り当て 9
 - 表示バッファサイズ 33
 - ページサイズ 11
- 非順次処理
 - アクセスメソッド 58
 - 予備ファイル 57
- 表示バッファ 33
 - サイズ 33
- ファクトテーブル
 - スタースキーマ結合 54
- プロシジャ
 - WHERE 処理, 変数名の変更 51
- ページサイズ
 - バッファ 11
- 変数
 - データセットの処理から削除する 16
 - データセットの処理時に保持する 30
 - 名前の変更 49
 - 変数名の変更 49
 - WHERE 処理時のプロシジャ 51
 - 出力時 50
 - 入力時 51

や

- 予備ファイル 57

ら

- ラベル
 - データセット 31
- ランダムアクセス
 - 予備ファイル 58

